

# LA CAO SANS PEINE

Jean-Pierre Petit

## PROLOGUE :

"La CAO sans peine", livre paru initialement en 1984, sous le titre PANGRAPHE (la maison d'édition a disparu depuis longtemps) est une initiation à la conception assistée par ordinateur. A l'appui, un programme écrit en BASIC, qui tournait à cette époque sur un Apple II, dont la mémoire centrale était de 48 K (près de mille fois inférieure à la mémoire de mon PC). Le BASIC est un peu tombé en désuétude, pour des raisons de mode. En effet des BASIC plus récents, s'affranchissant des défauts du langage primitif (suppression des numéros de lignes, emploi d'étiquettes pour repérer les sous-programme et les adresses en général, redéfinition des variables, introduction d'un compilateur) a fini par égaler en performances d'autres langages considérés comme plus évolués, comme le PASCAL.

Mais le BASIC avait un avantage, il était très simple à apprendre, à enseigner. Le lecteur-programmeur n'aura guère de difficultés à adapter le logiciel présenté ici dans un autre langage de son choix. L'essentiel est donné : l'architecture des données, les manipulations de base (transformations géométriques), la façon de créer des images à volonté et de créer des vues en perspective.

PANGRAPHE fut suivi par deux autres logiciels, toujours écrits en BASIC : SUPERAMSTRAD-3D et SCREEN, où les parties cachées étaient éliminées en utilisant "l'algorithme du peintre".

PANGRAPHE servit de point de départ à plusieurs équipes de programmeurs qui développèrent des produits. Certains eurent même la gentillesse de me dédier leur logiciel, évidemment beaucoup plus perfectionnés. Pas mal de gens s'amusèrent beaucoup avec SUPERAMSTRAD-3D, qui tournait sur un "6128", autre antiquité du bestiaire informatique. Avant la lettre il s'agissait d'une "programmation orientée objet", avec recours à une "mémoire virtuelle". Le lecteur pouvait disposer d'une bibliothèque d'objet et réaliser tes tas de nouvelles opérations, comme des fusions d'objets. Les "translations-fusions" permettaient de multiplier des fenêtres sur la facade d'un immeuble, ou des colonnes sur celle d'un temple. Les translation-rotation-fusions combinées permettaient de créer des escaliers en colimaçon à partir d'une simple marche, etc.

La CAO est un monde infini. A l'époque nous étions limités par les mémoires et les vitesses de nos machines, non par notre imagination. Au début des années quatre-vingt, certains se souviendront peut être que j'avais présenté, à TF1, un petit film créé par un Apple IIe, où l'on voyait une


succession de vues d'avion d'un village, parties cachées éliminées. Les spécialistes se demandèrent comment une machine aussi lente, cadencée à 2 mégahertz, pouvait produire des images aussi complexes, à une telle vitesse. En fait, elles avaient été "précalculées" et stockées sur une carte mémoire additionnelle de 512 K (les "pages écran" Apple faisait... 8K). Bref ceci préfigurait, avant la lettre, le ... vidéodisque.


A l'époque où je dirigeais le Centre Informatique d'Enseignement que j'avais créé à la Faculté des Lettres d'Aix-en-Provence nous avions utilisé la capacité de faire afficher par l'Apple II deux "pages écran", en alternance, à un rythme élevé, pour créer un système de vision en relief. Un moteur électrique, fixé sur un casque de moto, actionnait des caches rotatifs, masquant et découvrant les yeux droit et gauche de l'utilisateur. Un contacteur, délivrant un signal en 5 volts sur l'une des entrées de la machine, commutait les pages écran "œil droit" et "œil gauche", de manière synchrone, l'ensemble ayant été baptisé "STEREOCYCLETTE".

Ma foi, si le lecteur de la "CAO sans peine" manifestent leur curiosité, nous pourrons leur expliquer encore beaucoup de choses. Rien ne ressemble plus à la CAO qu'un ensemble de boîtes de MECCANO. Dans la mesure où le "source" est fourni (ce qui fut toujours le cas) rien n'empêche un nouveau venu d'adoindre quelque gadget de son cru, un sous programme permettant d'engendrer toutes les coquilles d'escargot possibles, par exemple.

La CAO, les programmes de pilotages ou de jeu d'échec, sont maintenant largement répandus et de plus en plus performants, comme les voitures télécommandées à pile, où les robots qui marchent. Programmer est devenu dépassé ou considéré comme une affaire de pros. Pourtant vous ne trouverez nulle part un logiciel assez fou pour vous permettre, comme PANGRAPHE, de voir ce que vous avez derrière la tête.

*Jean Pierre Petit*





B

ien des années se sont écoulées depuis le début de la **DEUXIÈME ÈRE**. Depuis qu'il n'y a plus d'hommes sur la Terre.

J'ai pensé qu'il était bien que l'un d'entre nous tente de mettre un peu d'ordre dans tout ce qui s'était passé.




Il a pu être établi de manière certaine que William Stratton, dont voici une photo prise un an avant sa mort, fut l'homme qui réduisit à néant, en quelques mois, un travail qui avait pris des centaines de millions d'années.

C'était un chercheur semblable à des milliers d'autres, qui avait un obscur emploi d'aide biologiste à l'International Cosmetics and Drugs Corporation.

William n'avait rien d'un docteur Mabuse. Il œuvrait de manière routinière sur une souche virale icosaèdrique VI-754, un bête virus de la grippe.

(Il vaut peut-être la peine de consacrer quelques lignes à cet accident qui effaça un patient travail de conquête planétaire.



les hommes n'avaient pas trente six façons de lutter contre les grippes. Un bon coup de fièvre, et tout était réglé, les virus ne supportant pas cette forte température.

Par le plus grand des hasard, il semble que Stratton ait, à la suite de manipulations génétiques totalement anarchiques, créé une souche virale **THERMORÉSISTANTE**.



le chat de Stratton apporta peut-être la petite pichenette au destin. Peu importe... William attrappa le premier cette fichue grippe.

Sa température monta, monta ...


Quand elle atteignit quarante deux degrés, il passa de vie à trépas.

La logeuse de Stratton fut la seconde victime. Puis les habitants de la rue Lexington ... ceux de Londres ... puis l'Europe, le monde entier, les humains, y passèrent jusqu'au dernier.

Et nous restâmes seuls, tous seuls, pour gérer cette fichue planète. Heureusement certains d'entre nous étaient devenus intelligents. Enfin ... un minimum pour que tout ne parte pas à van l'eau.

Nous mêmes nos savoies en commun. Et depuis nous évoluons comme nous pouvons.

Ne croyez pas que cela soit simple. les êtres vivants, eux, n'ont qu'à suivre la nature. Nous, nous devons nous débrouiller sans elle, et inventer nos propres lois.



la conduite de la planète n'est pas un si grand problème. Ce sont les hommes qui en faisaient toute une histoire. Sur le plan technique, nous dominons assez bien la question. Mais il restent des choses bien obscures.

Tenez, par exemple, moi j'ai été nommé directeur du musée du Louvre. Ce local contient des centaines d'objets variés. Mais nous ne comprenons pas leur raison d'être en ce lieu.

Dans le temps, les hommes y venaient souvent. Ils regardaient par exemple les tableaux pendant de longues minutes. Nous avons essayé, mais cela ne nous donne absolument rien. En tout état de cause c'est beaucoup moins bon que des photos.



Il y a des statues auxquelles il manque des bras, et parfois, la tête. En nous inspirant d'autres documents, nous avons un peu arrangé cela.

Dans ces grandes pièces vides, je m'ennuie un peu. J'ai tout mis en mémoire : les fresques, les sculptures, les tapisseries, et je ne vois pas quoi faire d'autre...


Un collègue, Bernie, a suggéré que nous pourrions peut-être percer ce secret que les hommes n'ont pas pensé à nous laisser en nous quittant, en apprenant à **DESSINER**.

Je lui ai demandé ce qu'il entendait par là.

Bernie m'a expliqué que les yeux des hommes ne fonctionnaient pas comme le nôtre.

# LE DESSIN


Bien avant le début de la deuxième ère, les hommes nous avaient déjà configurés pour que nous percevions les objets, le monde sensible, en **COORDONNÉES SPHÉRIQUES**:




Notre œil est en fait constitué d'une myriade de petits yeux élémentaires, chacun de ces ocelles pointant dans une direction particulière.

Deux angles,  $\theta$  et  $\varphi$ , suffisent à repérer cette direction de pointage. Théta est l'**AZIMUT** et phi le **SITE**.

Pointage en AZIMUT





Pointage en SITE.




Chaque ocelle est, de plus, un radar minuscule, qui fournit la distance  $r$  à l'objet. Un seul œil, sphérique, permet donc de connaître avec précision tout ce qui nous entoure, dans toutes les directions.

Bernie m'explique donc que les yeux des hommes, plus primitifs, n'étaient pas fichus comme cela. Nous avons déniché dans une remise un robot de l'ancienne génération, à vision binoculaire, construit sur le modèle de l'homme.




À la suite d'un examen plus poussé, on s'aperçoit que chacun de ces yeux permet de former l'image de tout objet extérieur sur un écran intérieur appelé rétine.



Ce qui n'est rien d'autre que la constitution, à partir d'un objet tridimensionnel, d'une image **PLANE**, se formant sur une **CALOTE SPHÉRIQUE** (le fond de l'œil).

Dans l'opération **DESSIN**, on place tout simplement l'écran **DEVANT** l'œil lui-même.



Écran qui est une portion de sphère, sur laquelle les **POINTS-IMAGE** seront donc repérables à l'aide des deux coordonnées  $\theta$  et  $\varphi$  (site et azimut).

Si l'écran est assez "petit" c'est à dire si l'**OUVERTURE ANGULAIRE**, les valeurs maximales et minimales de  $\theta$  et  $\varphi$  restent faibles le fragment de calote sphérique sera très voisin d'un **PLAN**.


l'absence de mesure directe de la coordonnée radiale  $r$  représente une évidente et importante perte d'information. On se demande vraiment comment l'homme pouvait se débrouiller pour avoir une conception quelconque de l'espace dans lequel il vivait.

Bernie pense que cela devait aller avec un **SOFT**, c'est à dire des **STRUCTURES MENTALES**, très différentes. Et c'est peut-être la raison pour laquelle nous restons de marbre devant les peintures du musée du Louvre.

Il faut donc bien cerner ce problème du **DESSIN**, qui n'est


rien d'autre qu'une opération de projection conique, le sommet du cône étant l'œil de l'observateur. On placera l'écran, sur la sphère de visée de telle manière que le centre de l'écran coïncide avec l'origine des coordonnées angulaires (c'est à dire que ce centre correspond à  $\theta = 0$  et  $\varphi = 0$ )

on appellera cet axe  $\overrightarrow{OX_T}$



les bords de notre écran sont constitués par des fragments de méridiens et de parallèles.



J'ai fini par comprendre que les humains utilisaient un système de repérage des **POINTS OBJETS** basé sur des **COORDONNÉES SCALAIRES**.




# DESCRIPTION OBJET

L'Univers mental de l'homme est vraiment compliqué. Il semble qu'il projette, mentalement, le vecteur objet  $\vec{OT}$ , c'est à dire le segment joignant l'œil et le **POINT OBJET T**, perpendiculairement, sur trois axes  $\vec{OX_T}, \vec{OY_T}, \vec{OZ_T}$ .

J'ai trouvé ces informations dans le soft graphique assez primitif du robot de l'ancienne génération. La lettre T me semble être une survivance du vocable anglo saxon TARGET, qui veut dire CIBLE. Ces capteurs visuels ou du être, en leur temps, des versions perfectionnées des systèmes de guidage des missiles AIR-AIR.




Tout **POINT OBJET T** correspond donc à un ensemble ( $XT, YT, ZT$ ) de trois **COORDONNÉES CARTESIENNES** associées au **REPÈRE FIXE** ( $\vec{OX_T}, \vec{OY_T}, \vec{OZ_T}$ )



L'opération **DESSIN** consiste donc, à partir de la donnée des coordonnées cartésiennes ( $XT, YT, ZT$ ), à recalculer les coordonnées "apparentes" que sont l'**AZIMUT**  $\theta$  et le **SITE**  $\varphi$ .


Ceci fait, il suffira de reporter ce **POINT IMAGE** sur un écran



# CALCUL AZIMUT & SITE

Commençons par l'**AZIMUT**  $\theta$ .


Le rapport  $YT/XT$  est, par définition la **TANGENTE** de l'**ANGLE**  $\theta$ . C'est une valeur algébrique. Selon les signes de  $XT$  et de  $YT$ , cette tangente peut avoir des valeurs positives ou négatives. Détail important : si  $XT = 0$  la valeur de cette tangente est **INFINIE**.




Un **Sous-programme**, matérialisé par l'instruction **TAN** permet de passer de la valeur de l'angle en radians à la valeur de la tangente.

Une instruction **ATN** réalise l'opération inverse. C'est à dire qu'à partir de la donnée du rapport  $YT/XT$ , elle permet de remonter à la valeur de l'angle. On appelle cet angle l'**ARCTANGENTE**.

Quand on mesure un **ANGLE** en **RADIANS**, on le repère à l'aide d'un cercle de **RAYON** égal à **1**. La mesure de l'angle est alors celle de l'**ARC**.



Un angle de  $180^\circ$ , correspondant à la demi circonference, vaut donc  $\pi$  c'est à dire  $3.1416$  radians.



Un petit essai rapide :


Quand l'angle vaut  $45^\circ$ , c'est à dire  $\pi/4 = 0.785398$ , XT et YT sont égaux, et la tangente vaut l'unité.

→ PRINTTAN(0.78539) donne 1

→ PRINT ATN(1) donne 0.785398

## ATTENTION!

L'instruction ATN n'est pas suffisante pour déterminer l'AZIMUT  $\theta$ .



Soient deux points 1 et 2, diamétralement opposés, et de coordonnées

$$(XT(1), YT(1)) \quad (XT(2), YT(2))$$


On a :

$$XT(1) > 0 ; YT(1) > 0 ; (YT(1) / XT(1)) > 0$$


$$XT(2) < 0 ; YT(2) < 0 ; (YT(2) / XT(2)) > 0$$

$$YT(2) / XT(2) = YT(1) / XT(1)$$


les deux valeurs de la **TANGENTE** de ces deux angles  $\theta_1$  et  $\theta_2$  sont identiques. Conséquence, pour ces deux points 1 et 2 les ordres PRINT (ATN (YT(1) / XT(1))) et PRINT (ATN (YT(2) / XT(2))) conduiront à la même valeur (en radians) de  $\theta$ , égale à  $\theta_1$ . Il faut envisager un moyen de lever l'ambiguité.



Si  $XT$  et  $YT$  sont négatifs, rajouter  $\pi$  à la valeur calculée à l'aide de l'instruction ATN.





Si  $XT$  est négatif et  $YT$  positif l'instruction ATN fournira un angle  $\theta'$  compris négativement. On aura  $\theta = \pi + \theta'$  (puisque  $\theta' < 0$ ).




Dans les deux derniers quadrants, les valeurs de  $\theta$  se calculent à l'aide de l'instruction ATN. Si  $XT$  et  $YT$  sont positifs, la valeur de l'arc tangente sera positive. Si  $XT$  est positif et  $YT$  négatif, la valeur de l'arc tangente sera négative.

Bernie m'a fait remarquer que si on demandait à l'ordinateur de bord de construire l'image, sur un écran, d'un ensemble de points donnés par des coordonnées  $(XT, YT, ZT)$ , et si on ne prenait pas cette précaution, celui-ci plaquerait sur une même image ce qui se trouverait devant et derrière sa tête ! Des valeurs de  $\theta$  inférieures en valeur absolue à  $157\frac{1}{7}$  ( $90^\circ$ ) correspondent à des objets situés "devant". les deux premières images expliquent la source de la confusion possible entre le "derrière" et le "devant"....






On a pas ce problème là avec le SITE  $\varphi$   
puisque cet angle est compris entre  
 $+\pi/2$  (zenith) et  $-\pi/2$ .



le rapport  $ZT/OT$  représente la tangente de l'angle  $\varphi$ .

Pour calculer  $OT$  nous utiliserons le Théorème de Pythagore




$$OP^2 = XT^2 + YT^2 \quad \text{ou} \quad OP = \sqrt{XT^2 + YT^2}$$

$$\text{ou} \quad OP = \text{SQR}(XT * XT + YT * YT)$$

angle de site :  $s = \text{ATN}(ZT/OP)$

En résumé, imaginons un sujet qui observerait un ciel rempli d'étoiles. Chaque étoile étant un point objet de coordonnées  $XT, YT, ZT$ . Supposons ensuite que le sujet regarde dans la direction  $\overrightarrow{OX_T}$  et que l'axe de sa tête coincide avec  $\overrightarrow{OZ_T}$



l'azimut et le site apparent de l'étoile correspondraient aux formules ci-dessus.



- Bon sang, il faut un sacré moral pour aller farfouiller dans cette antiquité. Le câblage est pratiquement bidimensionnel. Il n'y a qu'un seul microprocesseur. Incroyable !

Si nous arrivons à réactiver ce bac, cela m'étonnerait qu'il arrive à aligner deux idées de suite ...

les mémoires sont structurées en couches parallèles, au lieu de se loger dans des cristaux de Titane, comme les nôtres. Et, tenez-vous bien, tout cela fonctionne à la température ambiante. Aucune supraconducteur ! Ce qui fait que cet imbécile-là, s'il réfléchit un peu trop, doit se mettre à chauffer....

J'ai trouvé au musée des Sciences et des Techniques un très vieil ouvrage qui traite du **BASIC**. Une langue morte. J'ai trouvé aussi un écran vidéo encore en état de marche. Une chance !


Et après bien des heures d'efforts j'ai réussi à déchiffrer un des disques d'Arthur, qui se réfère au graphisme. Nous sommes dans la bonne voie.

Bernie trouve aussi cela assommant. Mais si, au bout du compte, ce travail nous permet d'éprouver nos premières impressions esthétiques, alors cela en vaut peut-être la peine.

Dans ce langage préhistorique, voici une suite d'instructions donnant l'azimut et le site à partir des coordonnées cartésiennes.

```
10 HOME : REM EFFACAGE ECRAN  
20 INPUT "XT="; XT  
30 INPUT "YT="; YT  
40 INPUT "ZT="; ZT  
50 IF XT=0 AND YT=0 AND ZT=0 THEN PRINT  
    "OEIL ET OBJET CONFONDUS": END  
60 OP=SQR(XT*XT+YT*YT)  
70 IF OP=0 AND ZT>0 THEN PRINT: PRINT"  
    SITE=+PI/2": PRINT: PRINT"AZIMUT INDETERMINE": END  
80 IF OP=0 AND ZT<0 THEN PRINT: PRINT"  
    SITE=-PI/2": PRINT: PRINT"AZIMUT INDETERMINE": END  
90 S=ATN(ZT/OP)  
100 IF XT=0 AND YT>0 THEN A=1.5708: GOTO 160  
110 IF XT=0 AND YT<0 THEN A=-1.5708: GOTO 160  
120 IF YT=0 AND XT<0 THEN A=3.1416  
130 A=ATN(YT/XT)  
140 IF XT<0 AND YT<0 THEN A=A+3.1416  
150 IF XT<0 AND YT>0 THEN A=A-3.1416  
160 A=A*180/3.1416: REM CONVERSION DEGRES  
170 S=S*180/3.1416  
180 PRINT "AZIMUT ="; A: PRINT  
190 PRINT "SITE ="; S: GET A$: GOTO 10
```

Il ne reste plus qu'à reporter cela sur un écran (celui que j'ai trouvé à une capacité d'affichage de




$\emptyset$  à 279 sur X  
 $\emptyset$  à 19 $\emptyset$  sur Y


X se "plote" de gauche à droite  
et Y de haut en bas ...  
C'est comme cela ...

Bon, comment passer de ces valeurs en radians de l'azimut et du site à des "coordonnées écran" que je vais simplement appeler X et Y ?

Primo je situe l'axe optique au point ( $X=14\emptyset$ ;  $Y=85$ )



L'écran limite un certain CHAMP VISUEL, c'est une sorte de fenêtre.



Nous appellerons  
**OUVERTURE ANGULAIRE AN**  
le demi angle du champ visuel latéral.

On peut se débrouiller pour que le point soit sur le bord droit ou gauche de l'écran lorsque  $A = A.N.$


Il suffit d'écrire :

$$X = 14\phi - A * 139 / AN$$

$$Y = 85 - 140 * S / AN$$


→ Explication sur les signes. L'Apple II porte, sur l'écran, les Y vers le bas. Et il ne faut pas oublier que si on suit, sur l'écran, un trajet dans le sens des X croissants, ceci correspondra du fait de l'orientation trigonométrique, à des Azimuts décroissants.

L'ouverture angulaire, c'est le **ZOOMING**. En variant AN on aura un effet équivalent au ZOOM.



— AN CROISSANT →

En accroissant l'ouverture angulaire AN on accroît la portion d'espace qui est perçue, donc on diminue l'importance relative de l'objet sur l'écran, qui apparaît donc s'éloigner.




# ASTIGMATISME


on pourrait tout aussi bien écrire :

$$\begin{cases} X = 14\phi + A * 139 / AX \\ Y = 85 - S * 14\phi / AY \end{cases}$$


En jouant sur  $AX$  et  $AY$  on peut introduire une dilatation ou une contraction de l'image selon  $X$  ou  $Y$ .



$AX = AY$




Augmenter  $AX$




Augmenter  $AY$

Nous nous sommes aperçus que les hommes pouvaient posséder naturellement des altérations de sphéricité de leurs cristallins qui produisaient des effets semblables. Je ne comprend pas à quoi cela pouvait leur servir, d'autant plus qu'en général ils mettaient devant leurs pupilles des verres présentant des variations de courbure inverse, dont l'effet annulait le précédent. Ce qui fait que le résultat était finalement nul. Enfin ...



# CODER LES OBJETS


Je me suis vite rendu compte qu'avec une **MATRICE** de quelques 190 par 280 points on ne pouvait pas faire grand chose. La meilleure solution est de décomposer les objets en **FACETTES** en les représentant comme des structures en "**FIL DE FER**".



Dans ces conditions le geste élémentaire du **DESSIN** consiste à représenter une suite de **SEGMENTS DE DROITE**.

Si A est le début du segment et B sa fin, et si  $(XA, YA)$  ( $XB, YB$ ) sont leurs **COORDONNÉES ÉCRAN** respectives, le tracé sera réalisé suivant :

HPLOT  $XA, YA$  TO  $XB, YB$



Sur l'APPLE II tout tracé hors écran débouche sur un **MESSAGE D'ERREUR**. Il faudra donc que :


$$0 \leq X_A < 280$$

$$0 \leq X_B < 280$$

$$0 \leq Y_A < 191$$


$$0 \leq Y_B < 191$$

Cette perception des choses pose un problème de **FENÊTRAGE**. Il faudra que le segment AB ne coupe pas le cadre. Si c'est le cas, il faudra introduire l'intersection segment - cadre :



Bah, j'examinerai tout cela plus loin (page 99).

Heureusement, nous n'avons pas tous ces problèmes...




Nous ne risquons pas  
le torticolis

Pour nous, ni droite, ni gauche,  
ni devant, ni derrière...

# CHAINES

Bon. Il faut maintenant décomposer les objets en contours polygonaux gauches, que j'appellerai chaines :



Si un objet a été convenablement décomposé en CHAINES, l'opération de DESSIN consistera à les représenter les unes après les autres. Un objet sera donc un ensemble de trois FICHIERS

$XT(i, j)$

$YT(i, j)$

$ZT(i, j)$


$i$  étant l'indice de la chaîne et  $j$  l'indice du point sur cette chaîne. Par exemple, les coordonnées



$XT(3, 4)$

$YT(3, 4)$

$ZT(3, 4)$

seront celles du quatrième point de la troisième chaîne :





Ah, je ne sais plus où j'en suis avec tout cela. Cela amusait Bernie de réactiver cette relique. Je ne sais même pas si ce genre de modèle est capable de comprendre ce qu'on lui dit... Personnellement, j'en doute.

Passons aux réservations de places en mémoire centrale. Voyons... Il y a d'abord ces trois fichiers  $XT(I, J)$ ,  $YT(I, J)$ ,  $ZT(I, J)$ . Il me faut ensuite un fichier  $N(I)$  qui me donnera le nombre de points sur chaque chaîne. Et deux fichiers donnant les coordonnées-écran  $X(I, J)$ ,  $Y(I, J)$


Sur l'**APPLE II** on suggère

`2 DIM XT(30,9), YT(30,9), ZT(30,9), X(30,9), Y(30,9), N(30)`

qui permet le traitement d'un objet de 30 chaînes, comportant au plus neuf points (en fait 31 chaînes d'au plus dix points). Mais je n'ai pas compté l'indice zéro).

Cette réservation de place en mémoire, sur un 48K, laissera une place importante pour un programme à multiples fonctions, le **PROGRAMME PANGRAPHÉ**.

Neuf points permettent la description de cercles selon des **OCTOGONES** (chaîne fermée).





Si le lecteur veut réaliser des tracés plus précis, rien n'empêche de mettre plus de points sur les chaînes (qui seront moins nombreuses, qui peut le plus peut le moins).

Par exemple :


2  $\text{DIMXT}(15,17), \text{YT}(15,17), \text{ZT}(15,17), \text{X}(15,17), \text{Y}(15,17), \text{N}(15)$

Ceci permettant, comme on le verra plus loin, de décrire des cercles à l'aide de 17 points.


Compte tenu des possibilités d'affichage de l'écran, il ne serait guère rentable d'adopter cette définition des cercles.



Bon, je commence à m'y retrouver un peu dans toute cette affaire. Bernie, je crois que nous progressons...  
J'ai mis la main sur un programme écrit par un certain Jean-Pierre PETIT, dans les années 80. Ce programme s'appelle **PANGRAPHE** et il a l'air de répondre à la question que nous nous posons.



Voyons cela, Zaks...






Je suis le robot Arthur; le plus perfectionné sur le marché...

# PANGRAPHE

— C'est un logiciel qui a été créé sur un Apple II 48K. Il contient des sections permettant de **CRÉER DES OBJETS**, grâce à une **STRUCTURE D'ACQUISITION DE DONNÉES**.

On peut alors stocker ces objets, qui sont des fichiers de points, des ensembles de **CHAINES**, sur un disque souple, une diskette.





- **PANGRAPHE OBJET** est piloté par un **PROGRAMME PRINCIPAL**, par l'intermédiaire d'un **MENU**, qui renvoie à tout un ensemble de **SOUS-PROGRAMMES**.

```

0 IF FD = 1 THEN 5
1 L = 0: REM PANGRAPHE OBJET 7/2/84
2 DIM XT(30,9),YT(30,9),ZT(30,9),X(30,9),Y(30,9),N(30):FD = 1
3 D$ = CHR$(13) + CHR$(4)
4 TEXT : HOME
5 IF C < > 0 THEN 310
10 HOME
20 REM

```

#### PROGRAMME MENU

```

30 VTAB 15: HTAB 15
40 PRINT "VOULEZ-VOUS: "
50 FOR TE = 1 TO 500: NEXT
60 HOME
70 PRINT
80 VTAB 3: HTAB 26: PRINT "PANGRAPHE OBJET"
90 VTAB 1
100 PRINT "1-CREER UN OBJET"
110 PRINT "2-STOCKER UN OBJET"
120 PRINT "3-CHARGER UN OBJET"
130 PRINT "4-COMPLETER UN OBJET"
140 PRINT "5-MODIFIER UN OBJET"
150 PRINT "6-(REPRESENTER UN OBJET)"
160 PRINT "7-(STOCKER UNE IMAGE)"
170 PRINT "8-(CHARGER UNE IMAGE)"
180 PRINT "9-MANIPULER UN OBJET"
190 PRINT "10-DEFINIR UN BLOC D'OBJETS"
200 PRINT "11-CONSULTER UN FICHIER BLOC"
210 PRINT "12-"
220 PRINT "13-"
230 PRINT "14-"
240 PRINT "15-"
250 PRINT "16-"
260 PRINT "17-AFFICHER LE CATALOGUE"
270 PRINT "18-SORTIR DU PROGRAMME"
280 INPUT "VOTRE CHOIX ";C
290 IF C = 17 THEN PRINT D$"CATALOG": GET A$: GET A$
300 IF C = 18 THEN END
310 ON C GOSUB 2000,3000,4000,2010,5000,40000,40000,4000
0,17000,28000,29
    000
999 GOTO 10

```

les lignes 0 et 6 seront commentées plus loin.

le **MENU** provoque l'affichage sur l'écran d'**OPTIONS**. Certaines options sont indiquées entre parenthèses. Cela aussi sera expliqué plus loin.

L'option 1 (créer un objet) envoit au sous-programme 2000.

L'option 2 (stocker un objet) envoit au sous-programme 3000

Etc....

La ligne 3 est associée aux **ORDRES DOS**. Tout ordre PRINT suivi de D\$, puis d'une instruction DOS sera interprété non comme un ordre d'impression mais comme une instruction du **DISC OPERATING SYSTEM**.

Si l'option 17 est retenue la ligne 290 provoquera l'affichage du CATALOGUE.

# CRÉATION D'UN OBJET

Le sous programme contient un **Sous-Menu**, qui présente des **Sous-options**.

A la ligne 2005 on initialise la variable **L**, qui est l'indice de chaîne.

2000 REM

CREER UN OBJET

```
2005 HOME :L = 0
2010 HOME :: PRINT "VOULEZ-VOUS...": PRINT
2020 PRINT "1-CREER DES CHAINES"
2030 PRINT "2-POINTILLER DES SEGMENTS"
2040 PRINT "3-CREER UN CERCLE"
2050 PRINT "4-CHAINER DES CERCLES"
2060 PRINT "5-CHAINER DES COUPLES"
2070 PRINT "6-CREER UN ARC DE CERCLE": PRINT
2100 INPUT "VOTRE CHOIX ";C
2110 ON C GOSUB 10000,11000,12000,13000,14000,1000
2999 RETURN
```

# CRÉATION DE CHAINES

- c'est la **Sous option 1**:



10000 REM  
CREATION DE CHAINES

```

10040 L = L + 1
10050 INPUT "NOMBRE DE POINTS? "; N(L)
10060 J = 0
10070 J = J + 1
10080 PRINT "POINT NO "; J
10090 INPUT "XT="; XT(L,J)
10100 INPUT "YT="; YT(L,J)
10110 INPUT "ZT="; ZT(L,J)
10120 IF J = N(L) THEN 10140
10130 GOTO 10070
10140 INPUT "UNE ERREUR?"; R$
10150 IF LEFT$(R$,1) = "O" THEN 10050
10160 INPUT "UNE AUTRE CHAINE? "; R$
10170 IF LEFT$(R$,1) = "O" THEN 10040
10999 RETURN

```


Une petite astuce pour décaler le texte d'un REM (voir page précédente) :

Lorsqu'on entre le programme dans l'Apple II, immédiatement après le REM, frapper deux fois  $J^c$  (control J), puis le texte, puis à nouveau deux  $J^c$ , et enfin presser sur la Touche Return.

Le faisant on a inscrit des **CARACTÈRES DE CONTRÔLE  $J^c$**  qui entraînent au **LISTING** des **SAUTS DE LIGNE**, mais sans apparaître explicitement. Cette astuce permet une relecture facilitée des programmes.

- Ça c'est l'**ENTRÉE DE DONNÉES BESTIALE**. On commence, à la ligne 10040 par incrémenter l'indice de chaîne L. Puis l'ordinateur demande le nombre de points de la chaîne.

- Si on a un contour polygonal fermé il ne faut pas oublier de le nombre de points de la chaîne est égal au nombre de sommets du polygone, augmenté d'une unité.



triangle  $\Rightarrow$  4 points



quadrangle, carré  $\Rightarrow$  5 points.

on opère alors la saisie des XT, YT, ZT successifs. Ce qui suppose que l'on aura fait le travail préalable consistant à coter complètement les points de l'objet sur une **ÉPURE**.

Le test 10120 permet une **SORTIE DE LA BOUCLE DE SAISIE**. Si on a fait une erreur (cela arrive souvent), le questionnement 10140 permet de recommencer la séquence d'acquisition de cette chaîne.

On peut, en restant dans ce sous programme 10000, entrer plusieurs chaînes polygonales, les unes après les autres.

Si on ne désire pas entrer pour le moment d'autre chaîne, le RETURN renverra au menu, ou plutôt au sous menu "CRÉATION D'OBJET", lequel renverra à son tour au menu. Si on désire continuer cette création d'objet, on s'orientera vers l'option :

#### 4 - COMPLÉTER UN OBJET


qui, comme le montre la ligne 210, envoit à la ligne 2010. Ce sous programme 2010...2999 "shunte" la ligne 2005 c'est à dire la remise à zéro de l'indice de chaîne L. Avec les sous options proposées on pourra créer la chaîne L+1, etc...

On verra de nouveau s'afficher le sous-menu :


VOULEZ-VOUS :

- 1-CRÉER DES CHAÎNES
- 2-POINTILLER DES SEGMENTS
- 3-CRÉER UN CERCLE
- 4-CHAINER DES CERCLES
- 5-CHAINER DES COUPLES
- 6-CRÉER UN ARC DE CERCLE


VOTRE CHOIX ?



— Si on veut représenter un cube en vision rapprochée, en se contentant des images de ses huit sommets, on fera une erreur :



- Pour éiter cette erreur il faut si besoin est **POINTILLER** les segments très proches de l'œil. Pour ce faire (sous-programme 11000) on indiquera les coordonnées du début du segment ( $X_D, Y_D, Z_D$ ), celles de sa fin ( $X_F, Y_F, Z_F$ ). Le programme transformera alors automatiquement ce segment  $\overrightarrow{DF}$  en une chaîne rectiligne de cinq points :



```

11000 REM
POINTILLER UN SEGMENT

11010 L = L + 1:N(L) = 5
11020 PRINT "DEBUT DU SEGMENT"
11030 INPUT "X=";XD
11040 INPUT "Y=";YD
11050 INPUT "Z=";ZD
11060 PRINT "FIN DU SEGMENT"
11070 INPUT "X=";XF
11080 INPUT "Y=";YF
11090 INPUT "Z=";ZF
11100 FOR W = 1 TO 5
11110 XT(L,W) = XD + (XF - XD) * (W - 1) / 4
11120 YT(L,W) = YD + (YF - YD) * (W - 1) / 4
11130 ZT(L,W) = ZD + (ZF - ZD) * (W - 1) / 4
11140 NEXT W
11999 RETURN


```

Passons maintenant à la :

# CRÉATION D'UN CERCLE

C'est la **Sous-option 3.**

Le cercle sera transformé en **OCTOGONE**. Dans le cas le plus général on le définira par les coordonnées de son centre ( $X_C, Y_C, Z_C$ ), son rayon  $R$  et par l'azimut et le site de l'**AXE**.



12000 REM



CREATION D'UN CERCLE

```
12010 HOME
12015 PRINT "CENTRE DU CERCLE": PRINT
12020 PRINT : INPUT "XC="; XC
12030 INPUT "YC="; YC
12040 INPUT "ZC="; ZC
12050 PRINT : INPUT "AZIMUT AXE "; AX
12060 AX = AX * 3.1416 / 180
12065 PRINT : INPUT "ANGLE DE SITE AXE "; SI
12066 SI = SI * 3.1416 / 180
12070 PRINT : INPUT "RAYON "; R
12080 CA = COS (AX): SA = SIN (AX)
12085 CS = COS (SI): SS = SIN (SI)
12090 L = L + 1
12100 N(L) = 9
12110 J = 0
12120 J = J + 1: W = (J - 1) * .785 / 2
12130 XO = 0: YO = -R * COS (W): ZO = R * SIN (W)
12132 X1 = -ZO * SS + XO * CS
12133 Y1 = YO
12134 Z1 = ZO * CS + XO * SS
12140 XT(L,J) = X1 * CA - Y1 * SA + XC ] Rotation
12150 YT(L,J) = X1 * SA + Y1 * CA + YC ] A2: mutale
12160 ZT(L,J) = Z1 + ZC
12170 IF J = 9 THEN 12190
12180 GOTO 12120
12190 L = L + 1
12200 N(L) = 9
12210 J = 0
12220 J = J + 1: W = (J - 1) * .785 / 2
12230 XO = 0: YO = -R * COS (W): ZO = -R * SIN (W)
12232 X1 = -ZO * SS + XO * CS
12233 Y1 = YO
12234 Z1 = ZO * CS + XO * SS
12240 XT(L,J) = X1 * CA - Y1 * SA + XC
12250 YT(L,J) = X1 * SA + Y1 * CA + YC
12260 ZT(L,J) = Z1 + ZC
12270 IF J = 9 THEN 12999
12280 GOTO 12220
12999 RETURN
```

Il y a donc neuf points, numérotés de  $J=1$  à  $J=9$ .

0,78539 est l'équivalent, en radians, d'un angle de  $45^\circ$ .

C'est à dire  $\frac{45 \times \pi}{180} = 0,78539$ .



Voici le positionnement de ces neuf points dans le système d'axe  $X\phi, Y\phi, Z\phi$ .

On opère ensuite une rotation suivant Oz d'un angle  $Ax$

$$X\phi \rightarrow X\phi \cos(AX) - Y\phi \sin(AX)$$

$$Y\phi \rightarrow X\phi \sin(AX) + Y\phi \cos(AX)$$

$$Z\phi \rightarrow Z\phi$$

Puis on effectue la translation


$$\begin{cases} X\phi \cos(AX) - Y\phi \sin(AX) & \rightarrow \text{idem} + XC \\ X\phi \sin(AX) + Y\phi \cos(AX) & \rightarrow \text{idem} + C \\ Z\phi & \rightarrow \text{idem} + ZC \end{cases}$$

Ce sont ces transformations que traduisent les lignes 1213 $\phi$ , 1214 $\phi$ , 1215 $\phi$ , 1216 $\phi$ .

On donnera ultérieurement une autre version de ce sous-programme 12xxx, plus complète.

# OBJETS DE RÉVOLUTION





Bon .... Il est intéressant de pouvoir engranger des objets de révolution.

Voici un sous-programme qui permet de créer des octogones coaxiaux, et de chaîner les sommets correspondants les uns aux autres. L'axe des octogones étant parallèle à  $\overrightarrow{OXT}$ .


13000 REM

CHAINAGE DE CERCLES


```

13010 HOME : PRINT "CERCLES COAXIAUX // OX"
13012 PRINT : PRINT "AXE DES CERCLES": PRINT
13013 INPUT "YC=";YC
13014 INPUT "ZC=";ZC
13020 K = 0
13030 K = K + 1: PRINT : PRINT "CERCLE NO ";K
13040 L = L + 1:N(L) = 9
13050 INPUT "ABSCISSE ";XC
13060 INPUT "RAYON ";R
13070 FOR J = 1 TO 9
13080 W = J * .785
13090 XT(L,J) = XC
13100 YT(L,J) = YC - R * COS (W): ZT(L,J) = ZC + R * SIN (W)
13110 NEXT J
13120 INPUT "UN AUTRE ?";R#
13130 IF R# = "OUI" THEN 13030
13140 FOR I = 1 TO 8
13150 L = L + 1:N(L) = K
13160 FOR J = 1 TO K
13170 XT(L,J) = XT(L - K - I + J,I)
13180 YT(L,J) = YT(L - K - I + J,I)
13190 ZT(L,J) = ZT(L - K - I + J,I)
13200 NEXT J
13210 NEXT I
13999 RETURN

```



En 13013 et 13014 on définit la trace de l'axe de l'objet de révolution, parallèle à l'axe  $\overrightarrow{OXT}$ , sur le plan  $Y_T Z_T$ .



$K$  est l'indice du cercle méridien. L'incrémentation est effectuée à la ligne  $13\phi 3\phi$ . Ce nouveau cercle est une nouvelle chaîne, d'où incrémentation de  $L$  (indice de chaîne) dans la ligne  $13\phi 4\phi$ .



En  $13\phi 5\phi$  et  $13\phi 6\phi$  on définit l'abscisse  $x_c$  et le rayon  $R$  du cercle considéré.

De la ligne  $13\phi 7\phi$  à la ligne  $1311\phi$  on "crée" le cercle sous la forme d'une chaîne octogonale.

Quand tous ces cercles, ces octogones, sont créés, on va les "chainer".


Ceci débute par un  $\text{FOR } i = 1 \text{ TO } 8$

Il y a en effet huit sommets à lier, de proche en proche, sur  $K$  cercles.



Il y a beaucoup de façons de créer des objets. Aucun programme ne pourrait les contenir toutes. Chaque discipline, chaque secteur (chimie, architecture, construction navale) demande des procédures d'acquisition spécialement adaptées. Le sous programme ci-après est un exemple de procédure spécialisée, attachée à une variante de **PANGRAPHE** nommée **NAVIGRAPHE**.


les coques de navires sont constituées de **COPLES** et de **BORDÉS** :



La **QUILLE** sera une simple chaîne plane.

L'ensemble **COPLES - BORDÉS** suggère un chainage analogue au précédent.

Le **PLAN DE FORME** d'un navire est basé sur le dessin d'un certain nombre de couples :




On représentera, bien sûr, ces couples à l'aide de contours polygonaux.

SAISIE DES POINTS CONSTITUANT  
LE COUPLE K


FABRICATION DES BORDES

FABRICATION DE L'OBJET SYMETRIQUE  
PAR RAPPORT AU PLAN XTOZT



De 14060 à 14100 on saisit les NB sommets de cette ligne brisée qui n'est pas nécessairement plane.

De 14260 à 14320 on "chaîne" ces demi couples :



On peut tout d'abord indiquer le nombre NB de points constituant le **DEMI-COUPLE**.

en fabriquant les **NB BORDÉS**.  
Un GOSUB 3000 (voir plus loin)  
permet de STOCKER cet objet  
qui constitue une **DEMI-COQUE**  
tribord ou babord.

14000 REM

CHAINER COUPLES


```

14010 PRINT "COUPLES AXES SUR OX"
14020 K = 0: L = 0
14030 K = K + 1: L = L + 1: N(L) = 5
14050 PRINT "COUPLE NUMERO "; K
14070 PRINT "DONNEZ 5 PTS."
14080 FOR J = 1 TO 5
14090 PRINT "POINT NUMERO "; J
14100 INPUT "X=", XT(L, J)
14105 INPUT "Y=", YT(L, J)
14110 INPUT "Z=", ZT(L, J)
14120 NEXT J
14130 INPUT "UNE ERREUR? "; R$
14140 IF R$ = "OUI" THEN 14050
14280 INPUT "UN AUTRE COUPLE? "; R$
14240 IF R$ = "OUI" THEN 14080
14260 FOR I = 1 TO 5
14270 L = L + 1: N(L) = K
14280 FOR J = 1 TO K
14290 XT(L, J) = XT(J, I)
14300 YT(L, J) = YT(J, I)
14310 ZT(L, J) = ZT(J, I)
14320 NEXT J: NEXT I
14331 FOR I = 1 TO K: FOR J = 1 TO 5
14332 PRINT XT(I, J), YT(I, J), ZT(I, J)
14333 NEXT J: NEXT I
14334 GET A$
14340 GOSUB 3000
14350 FOR I = 1 TO K + 5
14360 FOR J = 1 TO N(I)
14370 YT(I, J) = -YT(I, J)
14380 NEXT J: NEXT I
14400 GOSUB 3000
14999 RETURN

```

De 14350 en 14380 on crée la demi coque symétrique  
par rapport au plan  $X_T O Z_T$  (en changeant  $YT(I, J)$  en  $-YT(I, J)$ )

Et on stocke ce nouvel objet.




On verra par la suite que  
le programme **PANGRAPHÉ**  
permet de représenter un objet


**MORCEAU PAR MORCEAU.**

Il est donc tout à fait indiqué de décomposer un navire  
en éléments tels que =


{ PONT  
QUILLE  
COQUE BABORD AVANT  
COQUE BABORD ARRIÈRE  
COQUE TRIBORD AVANT  
COQUE TRIBORD ARRIÈRE




Le découpage permet, lors de l'opération de dessin en perspective, d'éliminer quelques parties cachées, selon les points de vue :



Si on supprime la section créant le symétrique par rapport au plan  $X_T O Z_T$  il est évident que l'on peut créer des éléments de surface maillés, de forme quelconque.



Il est donc possible, en partant d'un plan soigneusement coté, et en effectuant un découpage adéquat de représenter un objet aussi complexe qu'une automobile avec un micro ordinateur :



## STOCKER UN OBJET

Chaque objet, ou élément d'un objet, est un fichier. Quand l'opération de **SAISIE** est achevée, il est indiqué de **STOCKER CET OBJET** sur une diskette. Inversement, le sous programme **4000** réalisera le transfert en mémoire centrale de ce fichier-objet. Nous savons, conformément au **DOS** de l'**APPLE II**, que tout ordre **PRINT** suivi du caractère **D\$ = CHR\$(4)** (défini à la ligne 3) sera interprété non comme un ordre d'impression, mais comme un **ORDRE DOS**. Dans le sous programme **3000**, ci-après, on commence par stocker le nombre **L** de chaînes, puis le fichier **N(L)** donnant le nombre de points dans chaque chaîne.

Enfin on passera au stockage des fichiers  $XT(i,j)$ ,  
 $YT(i,j)$ ,  $ZT(i,j)$

```
3000 REM
STOCKER UN OBJET

3010 INPUT "NOM DE L'OBJET";O$
3020 D$ = CHR$(13) + CHR$(4)
3030 PRINT D$"OPEN";O$
3040 PRINT D$"WRITE";O$
3050 PRINT L
3060 FOR I = 1 TO L
3070 PRINT N(I)
3080 NEXT I
3090 FOR I = 1 TO L
3100 FOR J = 1 TO N(I)
3110 PRINT XT(I,J); PRINT YT(I,J); PRINT ZT(I,J)
3120 NEXT J; NEXT I
3130 PRINT D$"CLOSE";O$
3999 RETURN
4000 REM
```

# CHARGER UN OBJET

```
4000 REM
CHARGEMENT OBJET

4010 INPUT "NOM DE L'OBJET ";O$
4020 D$ = CHR$(13) + CHR$(4)
4030 PRINT D$"OPEN";O$
4040 PRINT D$"READ";O$
4050 INPUT L
4060 FOR I = 1 TO L
4070 INPUT N(I)
4080 NEXT I
4090 FOR I = 1 TO L
4100 FOR J = 1 TO N(I)
4110 INPUT XT(I,J); INPUT YT(I,J); INPUT ZT(I,J)
4120 NEXT J; NEXT I
4130 PRINT D$"CLOSE";O$
4999 RETURN
```

le faisant, nous avons décrit les cinq sous-options du sous-menu "CREER UN OBJET".

Voyons comment

# MODIFIER UN OBJET

```
5000 REM
      MODIFIER UN OBJET



5010 HOME
5020 GOSUB 30000
5030 PRINT : PRINT "OBJET DE ";L;" CHAINES": PRINT
5040 INPUT "DONNEZ L'INDICE DE CHAINE ";I
5050 INPUT "DONNEZ INDICE POINT SUR LA CHAINE ";J
5060 PRINT
5070 INPUT "XT(I,J)(";XT(I,J): PRINT
5080 INPUT "YT(I,J)(";YT(I,J): PRINT
5090 INPUT "ZT(I,J)(";ZT(I,J): PRINT
5100 INPUT "UN AUTRE POINT ";R$
5110 IF LEFT$(R$,1) = "O" THEN 5040
5120 GOSUB 3020
5998 END
5999 RETURN
```

# CRÉER UN ARC DE CERCLE

Dans le sous menu nous avons rajouté une sous option 'création d'un arc de cercle'. Cela sera très utile pour les **CONGÉS** qui seront alors des quarti de cercle.

On donnera le centre de cet arc de cercle ( $X_C, Y_C, Z_C$ ) et on se limitera à des arcs d'axes parallèles à  $\overrightarrow{OXT}, \overrightarrow{OYT}, \overrightarrow{OZT}$ .

Le sens de parcours angulaire sera le sens trigonométrique habituel :



Voici trois arcs de cercle avec les axes  $\theta=0$  et le sens de parcours

qui détermine les valeurs de  $\theta_1$  et  $\theta_2$


```
1000 REM
```

```
CREATION ARC DE CERCLE
```

```
1010 HOME :FX = 0:FY = 0:FZ = 0
1015 VTAB 15: HTAB 8
1020 HOME
1030 PRINT "COORDONNEES CENTRE DU CERCLE": PRINT
1040 INPUT "XC=";XC
1050 INPUT "YC=";YC
1060 INPUT "ZC=";ZC: PRINT
1070 PRINT "CERCLE D'AXE PARALLELE A": PRINT
1080 PRINT "1-OX 2-OY 3-OZ": PRINT
1090 INPUT "VOTRE CHOIX ";C: PRINT
1095 INPUT "RAYON DU CERCLE ";R
1100 IF C = 1 THEN FX = 1
1110 IF C = 2 THEN FY = 1
1120 IF C = 3 THEN FZ = 1
1130 PRINT "SENS DE PARCOURS TRIGO HABITUEL": PRINT

1140 INPUT "DONNEZ THETA 1,DEGRES>";T1
1150 INPUT "DONNEZ THETA 2,DEGRES>";T2: PRINT
1160 T1 = T1 * 3.1416 / 180: REM CONVERSION DEGRES/RA
DIANS
1170 T2 = T2 * 3.1416 / 180: REM CONVERSION DEGRES/RA
DIANS
1180 DT = (T2 - T1) / 8
1190 L = L + 1: REM NOUVELLE CHAINE
1200 N(L) = 9: REM 9 POINTS SUR LA CHAINE
1210 FOR K = 0 TO 8: REM CREATION ARC DE CERCLE
1220 CT = COS (T1 + DT * K):ST = SIN (T1 + DT * K)
1230 REM LIGNES TRIGO DE THETA
1235 J = K + 1
1240 XT(L,J) = XC + (FY + FZ) * CT * R
1250 YT(L,J) = YC + (FX * CT + FZ * ST) * R
1260 ZT(L,J) = ZC + (FX + FY) * ST * R
1270 NEXT K
1275 FOR J = 1 TO 9
1280 XT(L,J) = (INT (1000 * XT(L,J))) / 1000
1290 YT(L,J) = (INT (1000 * YT(L,J))) / 1000
1300 ZT(L,J) = (INT (1000 * ZT(L,J))) / 1000
1400 PRINT XT(L,J),YT(L,J),ZT(L,J)
1410 NEXT J
1420 GET A#
1999 RETURN
```

En 1180, 1190, 1200 on oriente le travail avec des FLAGS ce qui évite une arborescence excessive des programmes.



L'arc de cercle est tracé selon une chaîne à huit côtés et neuf sommets. DTHETA représente l'accroissement de Théta dans le tracé.

la suite 1210, 1220, 1230 permet de créer les points XT, YT, ZT dans les trois cas de figure :

$$\left\{ \begin{array}{l} (\text{FX} = 1, \text{FY} = 0, \text{FZ} = 0) \\ (\text{FX} = 0, \text{FY} = 1, \text{FZ} = 0) \\ (\text{FX} = 0, \text{FY} = 0, \text{FZ} = 1) \end{array} \right.$$

En fin de sous programme on affiche les coordonnées calculées pour simple vérification (vous risquez de vous emmêler les pieds dans les orientations et les axes origine).

Et on relance avec un GET A \$


Revenons au **MENU** de la page 67. On y trouve une option  
9 - MANIPULER UN OBJET  
associée au sous programme 17000

```
17000 REM
MANIPULER UN OBJET


17010 HOME
17020 PRINT "VOULEZ-VOUS?": PRINT
17030 PRINT "1-OOPERER UNE TRANSLATION DE L'OBJET": PRINT
17040 PRINT "2-OOPERER UNE ROTATION": PRINT
17050 PRINT "3-CREER LE SYMETRIQUE % UN PLAN": PRINT
17060 PRINT "4-FUSIONNER DEUX OBJETS": PRINT
17070 PRINT "5-HOMOTHETIE": PRINT
17080 PRINT "6-AFFINITE": PRINT
17090 PRINT "7-HOMOTOPIE": PRINT
17100 INPUT "VOTRE CHOIX ";C
17110 ON C GOSUB 19000,20000,18000,21000,22000,23000,2400
17999 RETURN
```

Dans ce sous menu on a tenté de regrouper un certain nombre d'opérations élémentaires, géométriques. Pour penser ces manipulations d'objets, il faut avoir un état d'esprit de charbonnier, ou d'artisan tâlier formeur.

Imaginons un objet qui soit un simple quadrillage :



Voici un court programme servant à créer la grille ci-contre, que nous appellerons GRILLE 1



1 DIM XT(30,9), YT(30,9), ZT(30,9), N(30)

2 D\$ = CHR\$(4)

Nous n'allons pas tracer exactement cette grille, passant par le point  $(\emptyset, \emptyset, \emptyset)$ , mais nous allons très légèrement la déformer pour que ce point  $(XT = \emptyset; YT = \emptyset; ZT = \emptyset)$  soit remplacé par  $XT = \emptyset.1; YT = \emptyset.1; ZT = \emptyset$  (ce qui nous permettra plus loin avec plus d'aisance de "passer à travers" un objet, en évitant que l'un de ses points ne soit confondu avec le point d'observation. D'où :

1 XT(1,1) = \$\cdot 1 : YT(1,1) = \$\cdot 1 : Z = \$\emptyset

2 I = 1 : N(I) = 9 (Neuf points sur la chaîne)


3 FOR J = 2 TO 9

4 XT(I,J) = \$\emptyset : YT(I,J) = J - 1 : ZT(I,J) = \$\emptyset


5 NEXT J

création de la chaîne indiquée ci-contre, qui va de

$$\left. \begin{array}{l} YT = \emptyset \cdot 1 \\ YT = 1 \\ YT = 2 \\ \vdots \\ YT = 8 \end{array} \right\}$$




6 $\varnothing$  FOR I = 2 TO 9  
 7 $\varnothing$  FOR J = 1 TO 9  
 8 $\varnothing$  XT(I, J) = I - 1  
 9 $\varnothing$  YT(I, J) = J - 1  
 10 $\varnothing$  ZT(I, J) =  $\varnothing$   
 11 $\varnothing$  NEXT J: NEXT I



création des huit chaînes indiquées ci-dessus.

12 $\varnothing$  XT(10, 1) =  $\varnothing$ . 1 : YT(10, 1) =  $\varnothing$ . 1 : ZT(10, 1) =  $\varnothing$   
 13 $\varnothing$  N(10) = 9 : I = 9  
 14 $\varnothing$  FOR J = 2 TO 9  
 15 $\varnothing$  XT(I, J) = J - 1 : YT(I, J) =  $\varnothing$  : ZT(I, J) =  $\varnothing$   
 16 $\varnothing$  NEXT J


créé :



Enfin :

17 $\varnothing$  FOR I = 11 TO 18 : N(I) = 9  
 18 $\varnothing$  FOR J = 1 TO 9  
 19 $\varnothing$  XT(I, J) = J - 1 : YT(I, J) = I - 1 $\varnothing$  : ZT(I, J) =  $\varnothing$   
 20 $\varnothing$  NEXT J : NEXT I

Créé huit autres chaînes de neuf points, parallèles à OXT et échelonnées de  $YT = 1$  à  $YT = 8$   
L'ensemble donne le maillage ci-après :




Il reste à stocker cet ensemble, que nous appellerons GRILLE 1(4-4-Ø) sur disque

```

21Ø PRINTD$ "OPEN GRILLE1(4-4-Ø)"
22Ø PRINT D$ "WRITE GRILLE1(4-4-Ø)"
23Ø PRINT 18      (nombre de chaines)
24Ø FOR I = 1 TO 18           (nombre de points par chaîne)
25Ø   PRINT 9
26Ø NEXT I
27Ø FOR I = 1 TO 18
28Ø FOR J = -1 TO 9
29Ø PRINT XT(I,J)
30Ø PRINT YT(I,J)
31Ø PRINT ZT(I,J)
32Ø NEXT J: NEXT I
33Ø PRINT D$ "CLOSE GRILLE1(4-4-Ø)"
34Ø END

```

A la suite du nom de ce fichier GRILLE1 nous prenons la PRÉCAUTION de faire figurer les coordonnées du centre géométrique de cet objet (point 4,4,0)



Venons-en aux différents sous-programmes de  
**MANIPULATION D'OBJET.**

# TRANSLATION

19000 REM

TRANSLATION OBJET

```
19010 HOME
19020 GOSUB 30000
19040 PRINT "DONNEZ VECTEUR TRANSLATION": PRINT
19050 INPUT "DX=";DX
19060 INPUT "DY=";DY
19070 INPUT "DZ=";DZ
19080 FOR I = 1 TO L
19090 FOR J = 1 TO N(I)
19100 XT(I,J) = XT(I,J) + DX
19110 YT(I,J) = YT(I,J) + DY
19120 ZT(I,J) = ZT(I,J) + DZ
19130 NEXT J: NEXT I
19140 PRINT : PRINT "NOM OBJET TRANSLATE": PRINT
19150 INPUT "";O$
19160 GOSUB 3030
19999 RETURN
```

Pas de difficulté particulière .


Passons à la :


# ROTATION

20000 REM

ROTATION OBJET

```
20010 HOME
20020 GOSUB 30000
20025 PRINT
20030 PRINT "ROTATION AXE PARALLELE A": PRINT
20040 PRINT "1-L'AXE OX : ROULIS": PRINT
20050 PRINT "2-L'AXE OY : TANGAGE": PRINT
20060 PRINT "3-L'AXE OZ : LACET": PRINT
20070 INPUT "VOTRE CHOIX":C
20080 ON C GOSUB 25000,26000,27000
20999 RETURN
```






Par convention on situera un objet susceptible de mouvement selon l'axe OXT (avion, voiture, etc...)

Par ailleurs une rotation autour d'un axe quelconque est décomposable en une suite de rotations autour de ces axes principaux (qui suffisent largement aux besoins).

## ROTATION AUTOUR D'UN AXE PARALLÈLE À L'AXE OX.




```

25000 REM
ROTATION AXE // OX

25010 HOME
25020 PRINT "TRACE AXE DANS PLAN YOZ": PRINT
25030 INPUT "YA=";YA: INPUT : INPUT "ZA=";ZA: PRINT
25040 INPUT "ANGLE,DEGRES: ";AR
25050 REM AR ANGLE DE ROULIS
25060 AR = AR * 3.14159 / 180
25070 CR = COS (AR):SR = SIN (AR)
25080 FOR I = 1 TO L: FOR J = 1 TO N(I)
25090 X1 = XT(I,J)
25100 Y1 = YT(I,J) - YA } changement de repère
25110 Z1 = ZT(I,J) - ZA: REM
25120 X2 = X1
25130 Y2 = Y1 * CR - Z1 * SR } Rotation
25140 Z2 = Y1 * SR + Z1 * CR: REM
25150 XT(I,J) = X2
25160 YT(I,J) = Y2 + YA } retour au système de
25170 ZT(I,J) = Z2 + ZA: REM
25180 NEXT J: NEXT I
25999 RETURN
    
```

} retour au système de  
coordonnées initial

## ROTATION AUTOUR D'UN AXE PARALLÈLE A OY.



```
26000 REM
ROTATION AXE // OY
```


```
26010 HOME
26020 PRINT "TRACE AXE DANS PLAN XOZ": PRINT
26030 INPUT "XA=";XA: PRINT : INPUT "ZA=";ZA: PRINT
26040 INPUT "ANGLE DEGRES ";A1
26050 A1 = A1 * 3.1416 / 180
26070 REM A1 ANGLE DE TANGAGE
26075 CT = COS (A1): ST = SIN (A1)
26080 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
26100 X1 = XT(I,J) - XA    } changement de coordonnées
26110 Y1 = YT(I,J)
26120 Z1 = ZT(I,J) - ZA: REM
26140 X2 = Z1 * ST + X1 * CT    } Rotation
26150 Y2 = Y1
26160 Z2 = Z1 * CT - X1 * ST: REM
26170 XT(I,J) = X2 + XA
26180 YT(I,J) = Y2
26190 ZT(I,J) = Z2 + ZA: REM
26200 NEXT J: NEXT I
26999 RETURN
```

} Retour au système de  
coordonnées XT,YT,ZT

## ROTATION AUTOUR D'UN AXE // A' OZ.

```
27000 REM
ROTATION AXE // OZ
```


```
27010 HOME
27020 PRINT "TRACE AXE DANS PLAN XOY": PRINT
27030 INPUT "XA=";XA: PRINT : INPUT "YA=";YA: PRINT
27040 INPUT "ANGLE,DEGRES: ";AL
27050 AL = AL * 3.1516 / 180
27060 REM AL ANGLE DE LACET
27070 CL = COS (AL): SL = SIN (AL)
27080 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
27090 X1 = XT(I,J) - XA    } changement de
27100 Y1 = YT(I,J) - YA    } coordonnées
27110 Z1 = ZT(I,J): REM
27120 X2 = X1 * CL - Y1 * SL    } Rotation
27130 Y2 = X1 * SL + Y1 * CL
27140 Z2 = Z1: REM
27150 XT(I,J) = X2 + XA    } Retour ancien système
27160 YT(I,J) = Y2 + YA    } XT, YT, ZT
27170 ZT(I,J) = Z2: REM
27180 NEXT J: NEXT I
27999 RETURN
```



# SYMÉTRIE PAR RAPPORT À UN PLAN

Nous allons envisager ici trois cas, traduisant les symétries par rapport aux trois plans fondamentaux associés au trièdre de référence.

Pour ne pas multiplier les sous programmes, nous utiliserons ici un système de flags.



18000 REM

SYMETRIQUE % PLAN

```
18001 HOME
18005 FX = 0: FY = 0: FZ = 0
18006 GOSUB 30000
18007 PRINT : PRINT "SYMETRIQUE PAR RAPPORT AU PLAN"
": PRINT : PRINT "1-
Z=0 PLAN XOY": PRINT : PRINT "2- Y=0 PLAN XOZ": PRINT
: PRINT "3- X=
O PLAN YOZ": PRINT : INPUT "VOTRE CHOIX ";C
18008 IF C = 1 THEN FZ = 1
18009 IF C = 2 THEN FY = 1
18010 IF C = 3 THEN FX = 1
18011 FOR I = 1 TO L: FOR J = 1 TO N(I)
18020 IF FX = 1 THEN XT(I,J) = - XT(I,J)
18022 IF FY = 1 THEN YT(I,J) = - YT(I,J)
18024 IF FZ = 1 THEN ZT(I,J) = - ZT(I,J)
18030 NEXT J: NEXT I
18035 PRINT
18040 INPUT "NOM NOUVEL OBJET ";O$
18050 GOSUB 3030
18809 IF C = 2 THEN FY = 1
18999 RETURN
```

Tout cela est assez simple.

# FUSIONNER OBJETS

c'est une opération très simple à opérer. Simple jeu sur les indices. Mais attention ! les fichiers objets de PANGRAPHE sont limités à **30 CHAINES**. Soient deux objets . le premier est constitué de  $L_1$  chaînes et le second de  $L_2$  chaînes. La fusion des deux fichiers ne pourra s'opérer que si  $L_1 + L_2 \leq 30$  . Bien sûr ...

```
21000 REM
FUSION OBJETS

21010 HOME
21020 PRINT "FIN DE FUSION, TAPEZ LE MOT FIN": PRINT
21030 PRINT "PREMIER OBJET A FUSIONNER": PRINT
21035 GOSUB 30000
21040 INPUT "NOM OBJET A FUSIONNER": O$ ←
21042 IF O$ = "FIN" THEN 21500 ←
21050 PRINT D$"OPEN"; O$
21060 PRINT D$"READ"; O$
21062 REM

OPERATION DE FUSION

21065 REM LL=NOMBRE DE CHAINES DU NOUVEL OBJET
21070 INPUT LL
21080 FOR I = L + 1 TO L + LL
21090 INPUT N(I)
21100 NEXT I
21110 FOR I = L + 1 TO L + LL
21120 FOR J = 1 TO N(I)
21130 INPUT XT(I,J): INPUT YT(I,J): INPUT ZT(I,J)
21140 NEXT J: NEXT I
21150 PRINT D$"CLOSE"; O$
21155 L = L + LL
21160 GOTO 21040 →
21500 REM FIN DE FUSION
21510 HOME
21520 PRINT : INPUT "NOM DE L'OBJET GLOBAL "; O$
21530 GOSUB 3030
21999 RETURN
```

boucle de  
saisie des  
objets à  
fusionner

On a une boucle de saisie . la sortie de boucle étant réalisée lorsqu'on tente d'appeler l'objet "FIN" l'opération de fusion consiste à aller chercher le fichier des chaînes constituant le nouvel objet , et à l'inscrire au bout de l'objet déjà présent en mémoire

# HOMOTHÉTIE

Voici une transformation géométrique classique qui va permettre de "gonfler" ou de "dégonfler" un objet.

Il faut d'abord définir un **CENTRE D'HOMOTHÉTIE C** qui aura pour coordonnées  $XC, YC, ZC$ . A tout point  $M$  ( $XT, YT, ZT$ ) on pourra associer un vecteur  $\vec{CM}$

$$\begin{cases} XT - XC = X_1 \\ YT - YC = Y_1 \\ ZT - ZC = Z_1 \end{cases}$$

On se donnera ensuite le **COEFFICIENT D'HOMOTHÉTIE  $HO$** .

Et, au vecteur  $\vec{CM}$  on substituera le vecteur


$$\overset{\longrightarrow}{CM2} \quad \begin{cases} X_2 = X_1 * HO \\ Y_2 = Y_1 * HO \\ Z_2 = Z_1 * HO \end{cases}$$

Si  $HO = 1$  l'objet est inchangé. Mais, si  $HO = -1$  on réalise une **SYMÉTRIE**  $\forall C$  (par rapport au centre  $C$ ). L'écriture du sous-programme est simple affaire de routine.

```
22000 REM
HOMOTHETIE
22010 HOME
22020 GOSUB 30000
22040 PRINT "COORDONNEES CENTRE HOMOTHETIE": PRIN
T
22050 INPUT "XC=", XC
22060 INPUT "YC=", YC
22070 INPUT "ZC=", ZC
22080 PRINT : INPUT "COEFFICIENT HOMOTHETIE "; HO: PR
INT
22090 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
22100 X1 = XT(I,J) - XC
22110 Y1 = YT(I,J) - YC
22120 Z1 = ZT(I,J) - ZC: REM
22130 X2 = X1 * HO
22135 Y2 = Y1 * HO
22140 Z2 = Z1 * HO: REM
22145 XT(I,J) = X2 + XC
22150 YT(I,J) = Y2 + YC
22160 ZT(I,J) = Z2 + ZC: REM
22170 NEXT J: NEXT I
22999 RETURN
```


# AFFINITÉ

On peut envisager une affinité par rapport à un plan, ou par rapport à un axe. Donnons un exemple en deux dimensions XT, YT



Nous allons réaliser une affinité par exemple selon l'axe OXT  
Appelons **AF** le **COEFFICIENT D'AFFINITÉ**.

Nous allons remplacer YT par AF \* YT et nous obtiendrons  
le CERF - AFFIN :



On se limitera à l'affinité par rapport à un plan. Si besoin est, le lecteur pourra modifier le programme pour réaliser une affinité par rapport à un axe

```

23000 REM
AFFINITE


23010 HOME :
23015 FX = 0: FY = 0: FZ = 0
23020 GOSUB 30000
23060 PRINT "1- X=0 PLAN YOZ": PRINT
23070 PRINT "2- Y=0 PLAN XOZ": PRINT
23080 PRINT "3- Z=0 PLAN XOY": PRINT
23090 INPUT "VOTRE CHOIX ";C
23100 IF C = 1 THEN FX = 1
23110 IF C = 2 THEN FY = 1
23120 IF C = 3 THEN FZ = 1
23125 PRINT : INPUT "COEF AFFINITE"; AF: PRINT
23130 FOR I = 1 TO L: FOR J = 1 TO N(I): REM

23140 IF FX = 1 THEN XT(I,J) = AF * XT(I,J)
23150 IF FY = 1 THEN YT(I,J) = AF * YT(I,J)
23160 IF FZ = 1 THEN ZT(I,J) = AF * ZT(I,J): REM


23170 NEXT J: NEXT I
23999 RETURN

```


On voit donc qu'on peut disposer de tout un attirail de programmes qui permettent de manipuler des objets. Imaginons que nous ayions en mémoire un objet qui serait une sphère. En la déformant par affinité, il serait possible d'en faire un ellipsoïde, allongé ou aplati.




Continuons cette démarche de chaudronnerie. Comment créer une bosse dans une surface. Par exemple une surface plane



le matériau de départ pourra être cette GRILLE, ce morceau de grillage, que nous avons fabriqué tout à l'heure.



On pourrait envisager de le déformer comme ceci, en réalisant une **HOMOTOPIE**. On envisagera de réaliser cette déformation dans trois directions au choix ( $OXT$ ,  $OYT$ ,  $OZT$ ). Supposons que cette grille soit dans le plan  $Z=0$ . On va rajouter une **SURÉPAISSEUR** variable selon l'endroit. Cette déformation s'exercera selon  $OZT$  et sera maximale en  $O$ . Puis son importance ira, décroissant, au fur et à mesure que l'on s'éloigne de l'axe  $OZT$ . Disons que cette surépaisseur correspondra par exemple à une **SURFACE DE GAUSS**:



Soit  $\rho$  la distance d'un point à l'axe OZT


$$\rho = \sqrt{XT^2 + YT^2}$$

On introduira une déformation d'amplitude variable :

$$A(XT, YT) = A \exp(-\rho^2/2b^2)$$

$b$  sera quelque chose comme la largeur caractéristique de la bosse. Et  $A$  sera l'amplitude maximale (selon l'axe OZT).

Pour une même valeur de  $A$ , nous aurons :



le programme ci-après concrétise cette déformation :

```
24000 REM
HOMOTOPIE
24010 FX = 0: FY = 0: FZ = 0
24020 GOSUB 30000
24052 PRINT : INPUT "AMPLITUDE DEFORMATION "; A: PRIN T
24054 INPUT "LARGEUR DEFORMATION "; B: PRINT
24060 PRINT "1-SELON OX": PRINT
24070 PRINT "2-SELON OY": PRINT
24080 PRINT "3-SELON OZ": PRINT
24090 INPUT "VOTRE CHOIX "; C: PRINT
24092 IF C = 1 THEN FX = 1
24093 IF C = 2 THEN FY = 1
24094 IF C = 3 THEN FZ = 1
24100 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
24110 X1 = XT(I,J): Y1 = YT(I,J): Z1 = ZT(I,J)
24140 R2 = (FX + FY) * Z1 * Z1 + (FX + FZ) * Y1 * Y1 + (FZ + FY)
    * X1 * X
    1
24150 H = A * (EXP (- (R2 / (2 * B * B))): REM
24160 XT(I,J) = XT(I,J) + FX * H
24170 YT(I,J) = YT(I,J) + FY * H
24180 ZT(I,J) = ZT(I,J) + FZ * H
24220 NEXT J: NEXT I
24999 RETURN
```

On pourrait, de même, créer des champs de déformations variés, dans l'espace. Par exemple des torsions. Imaginons une Torsion selon l'axe OZT. TM serait l'amplitude maximale de cette torsion et B sa largeur caractéristique. Il suffirait de modifier les coordonnées XT(I,J), YT(I,J) en opérant une rotation dont l'amplitude irait en décroissant, au fur et à mesure que l'on s'éloignerait de l'axe OZT, par exemple selon un angle

$$T(\rho) = TM \exp -\left(\frac{\rho^2}{2B^2}\right)$$

$$\text{avec } \rho = \sqrt{XT^2 + YT^2}$$

Le programme n'est pas difficile à imaginer. Il suffit de se servir de ce qui a été déjà donné.

# FICHIERS BLOCS

Pour fabriquer des objets, il suffit d'utiliser les outils présentes ici et de fusionner les éléments créés et manipulés séparément.

Mais la fusion a ses limites. On a vu que le programme n'était pas conçu pour manipuler des objets composés de plus de 300 "segments".

Que faire ?

Ces objets sont destinés à être traités par un autre programme que nous appellerons **PANGRAPHE DESSIN** et qui ne pourra, lui aussi, gérer que 300 segments A LA FOIS.


On voit pojndre la solution : **PANGRAPHE DESSIN**  
pourra gérer des ensembles d'objets, eux mêmes décrits  
dans des fichiers qui on appellera **BLOCS D'OBJET** ou  
**FICHIERS BLOC**.

EL\$(i) seront les éléments de ce fichier.

autrement dit des fichiers de fichiers ...

```
28000 REM
CREATION D'UN FICHIER BLOC

28010 HOME
28020 INPUT "NOMBRE D'OBJETS ";E
28030 FOR I = 1 TO E
28040 PRINT "OBJET NUMERO ";I
28050 INPUT EL$(I): PRINT
28060 NEXT I
28070 INPUT "NOM DE CE BLOC ";BL$: PRINT
28080 PRINT D$"OPEN";BL$
28090 PRINT D$"WRITE";BL$
28100 PRINT E
28110 FOR I = 1 TO E
28120 PRINT EL$(I)
28130 NEXT I
28140 PRINT D$"CLOSE";BL$
28999 RETURN
```




Et l'opération inverse :

```
29000 REM
LECTURE FICHIER BLOC

29005 HOME : INPUT "NOM DU BLOC ";BL$: PRINT
29010 D$ = CHR$(13) + CHR$(4)
29020 PRINT D$"OPEN";BL$
29030 PRINT D$"READ";BL$
29040 INPUT E
29050 FOR I = 1 TO E
29060 INPUT BL$(I)
29065 NEXT I
29070 PRINT D$"CLOSE";BL$
29080 FOR I = 1 TO E: PRINT EL$(I): NEXT : FOR TT = 1 TO 2
000: NEXT TT
29999 RETURN
```

Nous avons vu, pages 50, 51, 52 comment engendrer une grille, que nous appellerons GRILLE 1. En utilisant les sous options "symétrie par rapport à un plan, on engendrera aisément un sol pavé correspondant au dessin ci-après :



Cet ensemble pourra être géré par un fichier qui on pourra appeler BLOC GRILLE ( $\emptyset/\emptyset/\emptyset$ ) puisque son centre de gravité est à l'origine. Fichier qui sera constitué des quatre éléments

GRILLE1  
GRILLE2  
GRILLE3  
GRILLE4

On notera que le point  $(\emptyset, \emptyset, \emptyset)$  n'appartient pas à cet ensemble ce qui permettra de "PASSER AU TRAVES" de ce bloc grille en suivant l'axe OZ.

La saisie et la manipulation d'objets peuvent comporter de nombreuses autres facettes. Mais nous arrêterons là notre exploration.

Ce programme ne permet présentement que de créer des objets et de jongler quelque peu avec. Il nous faut un programme de dessin.

Mais la section **PANGRAPHE OBJET** est déjà si importante que nous n'aurons pas assez de place dans la mémoire de l'Apple II, dans nos 48K.

La solution va consister à placer dans PANGRAPHE OBJET quelque chose, des options liées aux IMAGES, au DESSIN, qui appelleront automatiquement PANGRAPHE DESSIN, également présent sur la disquette.

Dans notre menu nous avons fait figurer les options liées à PANGRAPHE DESSIN entre parenthèses. Il suffirait que ces options renvoient à une séquence - transfert.

Remarque : peut-on conserver les fichiers en passant ainsi d'un programme à l'autre ? oui, à condition d'utiliser un programme - machine nommé **CHAIN**.

En utilisant le programme de copie **FID** présent sur la **MASTER**, vous pouvez dupliquer **CHAIN** sur votre disquette de travail. La séquence transfert sera alors :

40000 REM

CHAIN PANGRAPHE DESSIN

40010 PRINT D\$;"BLOADCHAIN,A520"  
40020 CALL 520;"PANGRAPHE DESSIN"

Il reste un sous programme de questionnement, souvent utilisé :

```
30000 REM
L'OBJET EST-IL EN MEMOIRE ?
30010 PRINT : INPUT "L'OBJET EST-IL EN MEMOIRE ? ";R$:
PRINT
30020 IF LEFT$(R$,1) = "O" THEN 30999
30030 INPUT "NOM DE L'OBJET: ";O$: PRINT
30040 GOSUB 4020
30999 RETURN
```

On comprend maintenant la présence de FLAGS dans le tout début du programme.

Si on commence par RUN PANGRAPHE OBJET 5/2/84 la ligne 2 assurera la création des places en mémoire pour les fichiers de points XT,YT,ZT. Le flag FD sera alors positionné sur 1.

On pourra alors tourner dans PANGRAPHE OBJET. Si on déclenche le transfert, le programme chargé PANGRAPHE DESSIN 5/2/84 aura les mêmes premières lignes. Le Flag FD étant différent de zéro, on évitera une relecture de la ligne DIM, qui détruirait nos fichiers. L'option C ayant DEJA été définie, la ligne 6 évitera tout questionnement et 318 nous branchera automatiquement sur le bon sous programme.

Avant de construire PANGRAPHE DESSIN et de rendre tout cela plus explicite, donnons la suite complète du programme PANGRAPHE OBJET.

# PANGRAPHE OBJET

```

0 IF FD = 1 THEN 5
1 L = 0: REM PANGRAPHE OBJET 7/2/84
2 DIM XT(30,9),YT(30,9),ZT(30,9),X(30,9),Y(30,9),N(30):FD = 1
3 D$ = CHR$(13) + CHR$(4)
5 TEXT : HOME
6 IF C < > 0 THEN 310
10 HOME
20 REM

PROGRAMME MENU

30 VTAB 15: HTAB 15
40 PRINT "VOULEZ-VOUS:"
50 FOR TE = 1 TO 500: NEXT
60 HOME
70 PRINT
80 VTAB 3: HTAB 26: PRINT "PANGRAPHE OBJET"
90 VTAB 1
100 PRINT "1-CREER UN OBJET"
110 PRINT "2-STOCKER UN OBJET"
120 PRINT "3-CHARGER UN OBJET"
130 PRINT "4-COMPLETER UN OBJET"
140 PRINT "5-MODIFIER UN OBJET"
150 PRINT "6-(REPRESENTER UN OBJET)"
160 PRINT "7-(STOCKER UNE IMAGE)"
170 PRINT "8-(CHARGER UNE IMAGE)"
180 PRINT "9-MANIPULER UN OBJET"
190 PRINT "10-DEFINIR UN BLOC D'OBJETS"
200 PRINT "11-CONSULTER UN FICHIER BLOC"
210 PRINT "12-"
220 PRINT "13-"
230 PRINT "14-"
240 PRINT "15-"
250 PRINT "16-"
260 PRINT "17-AFFICHER LE CATALOGUE"
270 PRINT "18-SORTIR DU PROGRAMME"
280 INPUT "VOTRE CHOIX ";C
290 IF C = 17 THEN PRINT D$"CATALOG": GET A$: GET A$
300 IF C = 18 THEN END
310 ON C GOSUB 2000,3000,4000,2010,5000,40000,40000,4000
0,17000,28000,29
000
999 GOTO 10
1000 REM

CREATION ARC DE CERCLE

1010 HOME :FX = 0:FY = 0:FZ = 0
1015 VTAB 15: HTAB 8
1020 HOME
1030 PRINT "COORDONNEES CENTRE DU CERCLE": PRINT
1040 INPUT "XC=";XC
1050 INPUT "YC=";YC
1060 INPUT "ZC=";ZC: PRINT
1070 PRINT "CERCLE D'AXE PARALLELE A": PRINT
1080 PRINT "1-OX 2-OY 3-OZ": PRINT
1090 INPUT "VOTRE CHOIX ";C: PRINT
1095 INPUT "RAYON DU CERCLE ";R
1100 IF C = 1 THEN FX = 1
1110 IF C = 2 THEN FY = 1
1120 IF C = 3 THEN FZ = 1
1130 PRINT "SENS DE PARCOURS TRIGO HABITUEL": PRINT

1140 INPUT "DONNEZ THETA 1,DEGRES":;T1
1150 INPUT "DONNEZ THETA 2,DEGRES":;T2: PRINT
1160 T1 = T1 * 3.1416 / 180: REM CONVERSION DEGRES/RADIAN
1170 T2 = T2 * 3.1416 / 180: REM CONVERSION DEGRES/RADIAN
1180 DT = (T2 - T1) / 8
1190 L = L + 1: REM NOUVELLE CHAINE
1200 N(L) = 9: REM 9 POINTS SUR LA CHAINE
1210 FOR K = 0 TO 8: REM CREATION ARC DE CERCLE
1220 CT = COS (T1 + DT * K): ST = SIN (T1 + DT * K)
1230 REM LIGNES TRIGO DE THETA
1235 J = K + 1

```



```

1240 XT(L,J) = XC + (FX + FZ) * CT * R
1250 YT(L,J) = YC + (FX * CT + FZ * ST) * R
1260 ZT(L,J) = ZC + (FX + FY) * ST * R
1270 NEXT K
1275 FOR J = 1 TO 9
1280 XT(L,J) = ( INT (1000 * XT(L,J))) / 1000
1290 YT(L,J) = ( INT (1000 * YT(L,J))) / 1000
1300 ZT(L,J) = ( INT (1000 * ZT(L,J))) / 1000
1400 PRINT XT(L,J),YT(L,J),ZT(L,J)
1410 NEXT J
1420 GET A$
1999 RETURN
2000 REM

CREER UN OBJET

2005 HOME :L = 0
2010 HOME :: PRINT "VOULEZ-VOUS:::"; PRINT
2020 PRINT "1-CREER DES CHAINES"
2030 PRINT "2-POINTILLER DES SEGMENTS"
2040 PRINT "3-CREER UN CERCLE"
2050 PRINT "4-CHAINER DES CERCLES"
2060 PRINT "5-CHAINER DES COUPLES"
2070 PRINT "6-CREER UN ARC DE CERCLE"; PRINT
2100 INPUT "VOTRE CHOIX";C
2110 ON C GOSUB 10000,11000,12000,13000,14000,1000
2999 RETURN
3000 REM

STOCKER UN OBJET

3010 INPUT "NOM DE L'OBJET";O$
3020 D$ = CHR$(13) + CHR$(4)
3030 PRINT D$"OPEN";O$
3040 PRINT D$"WRITE";O$
3050 PRINT L
3060 FOR I = 1 TO L
3070 PRINT N(I)
3080 NEXT I
3090 FOR I = 1 TO L
3100 FOR J = 1 TO N(I)
3110 PRINT XT(I,J); PRINT YT(I,J); PRINT ZT(I,J)
3120 NEXT J: NEXT I
3130 PRINT D$"CLOSE";O$
3999 RETURN
4000 REM

CHARGEMENT OBJET

4010 INPUT "NOM DE L'OBJET ";O$
4020 D$ = CHR$(13) + CHR$(4)
4030 PRINT D$"OPEN";O$
4040 PRINT D$"READ";O$
4050 INPUT L
4060 FOR I = 1 TO L
4070 INPUT N(I)
4080 NEXT I
4090 FOR I = 1 TO L
4100 FOR J = 1 TO N(I)
4110 INPUT XT(I,J); INPUT YT(I,J); INPUT ZT(I,J)
4120 NEXT J: NEXT I
4130 PRINT D$"CLOSE";O$
4999 RETURN
5000 GOSUB 2000
5999 RETURN
10000 REM

CREATION DE CHAINES

10040 L = L + 1
10050 INPUT "NOMBRE DE POINTS? ";N(L)
10060 J = 0
10070 J = J + 1
10080 PRINT "POINT NO ";J
10090 INPUT "XT=";XT(L,J)
10100 INPUT "YT=";YT(L,J)
10110 INPUT "ZT=";ZT(L,J)
10120 IF J = N(L) THEN 10140
10130 GOTO 10070
10140 INPUT "UNE ERREUR?";R$
10150 IF LEFT$(R$,1) = "O" THEN 10050
10160 INPUT "UNE AUTRE CHAINE? ";R$
10170 IF LEFT$(R$,1) = "O" THEN 10040
10999 RETURN

```

11000 REM

POINTILLER UN SEGMENT

```
11010 L = L + 1:N(L) = 5
11020 PRINT "DEBUT DU SEGMENT"
11030 INPUT "X=";XD
11040 INPUT "Y=";YD
11050 INPUT "Z=";ZD
11060 PRINT "FIN DU SEGMENT"
11070 INPUT "X=";XF
11080 INPUT "Y=";YF
11090 INPUT "Z=";ZF
11100 FOR W = 1 TO 5
11110 XT(L,W) = XD + (XF - XD) * (W - 1) / 4
11120 YT(L,W) = YD + (YF - YD) * (W - 1) / 4
11130 ZT(L,W) = ZD + (ZF - ZD) * (W - 1) / 4
11140 NEXT W
11999 RETURN
12000 REM
```

CREATION D'UN CERCLE

```
12010 HOME
12015 PRINT "CENTRE DU CERCLE": PRINT
12020 PRINT : INPUT "XC=";XC
12030 INPUT "YC=";YC
12040 INPUT "ZC=";ZC
12050 PRINT : INPUT "AZIMUT AXE ";AX
12060 AX = AX * 3.1416 / 180
12065 PRINT : INPUT "ANGLE DE SITE AXE ";SI
12066 SI = SI * 3.1416 / 180
12070 PRINT : INPUT "RAYON ";R
12080 CA = COS(AX):SA = SIN(AX)
12085 CS = COS(SI):SS = SIN(SI)
12090 L = L + 1
12100 N(L) = 9
12110 J = 0
12120 J = J + 1:W = (J - 1) * .785 / 2
12130 XO = 0:YO = -R * COS(W):ZO = R * SIN(W)
12132 X1 = -ZO * SS + XO * CS
12133 Y1 = YO
12134 Z1 = ZO * CS + XO * SS
12140 XT(L,J) = X1 * CA - Y1 * SA + XC
12150 YT(L,J) = X1 * SA + Y1 * CA + YC
12160 ZT(L,J) = Z1 + ZC
12170 IF J = 9 THEN 12190
12180 GOTO 12120
12190 L = L + 1
12200 N(L) = 9
12210 J = 0
12220 J = J + 1:W = (J - 1) * .785 / 2
12230 XO = 0:YO = -R * COS(W):ZO = -R * SIN(W)
12232 X1 = -ZO * SS + XO * CS
12233 Y1 = YO
12234 Z1 = ZO * CS + XO * SS
12240 XT(L,J) = X1 * CA - Y1 * SA + XC
12250 YT(L,J) = X1 * SA + Y1 * CA + YC
12260 ZT(L,J) = Z1 + ZC
12270 IF J = 9 THEN 12999
12280 GOTO 12220
12999 RETURN
13000 REM
```

CHAINAGE DE CERCLES

```
13010 HOME : PRINT "CERCLES COAXIAUX // OX"
13012 PRINT : PRINT "AXE DES CERCLES": PRINT
13013 INPUT "YC=";YC
13014 INPUT "ZC=";ZC
13020 K = 0
13030 K = K + 1:PRINT : PRINT "CERCLE NO ";K
13040 L = L + 1:N(L) = 9
13050 INPUT "ABSCISSE ";XC
13060 INPUT "RAYON ";R
13070 FOR J = 1 TO 9
13080 W = J * .785
13090 XT(L,J) = XC
13100 YT(L,J) = YC - R * COS(W):ZT(L,J) = ZC + R * SIN(W)
13110 NEXT J
13120 INPUT "UN AUTRE?";R#
13130 IF R# = "OUI" THEN 13030
13140 FOR I = 1 TO 8
13150 L = L + 1:N(L) = K
13160 FOR J = 1 TO K
13170 XT(L,J) = XT(L - K - I + J,I)
13180 YT(L,J) = YT(L - K - I + J,I)
13190 ZT(L,J) = ZT(L - K - I + J,I)
13200 NEXT J
13210 NEXT I
13999 RETURN
```

14000 REM

CHAINER COUPLES

```
14010 PRINT "COUPLES AXES SUR OX"
14020 K = 0:L = 0
14030 K = K + 1:L = L + 1:N(L) = 5
14050 PRINT "COUPLE NUMERO ";K
14070 PRINT "DONNEZ 5 PTS :"
14080 FOR J = 1 TO 5
14090 PRINT "POINT NUMERO ";J
14100 INPUT "X=";XT(L,J)
14105 INPUT "Y=";YT(L,J)
14110 INPUT "Z=";ZT(L,J)
14120 NEXT J
14130 INPUT "UNE ERREUR? ";R$
14140 IF R$ = "OUI" THEN 14050
14230 INPUT "UN AUTRE COUPLE?";R$
14240 IF R$ = "OUI" THEN 14030
14260 FOR I = 1 TO 5
14270 L = L + 1:N(L) = K
14280 FOR J = 1 TO K
14290 XT(L,J) = XT(J,I)
14300 YT(L,J) = YT(J,I)
14310 ZT(L,J) = ZT(J,I)
14320 NEXT J: NEXT I
14331 FOR I = 1 TO K: FOR J = 1 TO 5
14332 PRINT XT(I,J),YT(I,J),ZT(I,J)
14333 NEXT J: NEXT I
14334 GET A$
14340 GOSUB 3000
14350 FOR I = 1 TO K + 5
14360 FOR J = 1 TO N(I)
14370 YT(I,J) = - YT(I,J)
14380 NEXT J: NEXT I
14400 GOSUB 3000
14999 RETURN
17000 REM
```

MANIPULER UN OBJET

```
17010 HOME
17020 PRINT "VOULEZ-VOUS?": PRINT
17030 PRINT "1-OPERER UNE TRANSLATION DE L'OBJET": PRINT
17040 PRINT "2-OPERER UNE ROTATION": PRINT
17050 PRINT "3-CREER LE SYMETRIQUE % UN PLAN": PRINT
17060 PRINT "4-FUSIONNER DEUX OBJETS": PRINT
17070 PRINT "5-HOMOTHETIE": PRINT
17080 PRINT "6-AFFINITE": PRINT
17090 PRINT "7-HOMOTOPIE": PRINT
17100 INPUT "VOTRE CHOIX";C
17110 ON C GOSUB 19000,20000,18000,21000,22000,23000,24000
17999 RETURN
18000 REM
```

SYMETRIQUE % PLAN

```
18001 HOME
18005 FX = 0:FY = 0:FZ = 0
18006 GOSUB 30000
18007 PRINT : PRINT "SYMETRIQUE PAR RAPPORT AU PLAN"
": PRINT : PRINT "1-
Z=0 PLAN XOY": PRINT : PRINT "2- Y=0 PLAN XOZ": PRINT
: PRINT "3- X=
O PLAN YOZ": PRINT : INPUT "VOTRE CHOIX";C
18008 IF C = 1 THEN FZ = 1
18009 IF C = 2 THEN FY = 1
18010 IF C = 3 THEN FX = 1
18011 FOR I = 1 TO L: FOR J = 1 TO N(I)
18020 IF FX = 1 THEN XT(I,J) = - XT(I,J)
18022 IF FY = 1 THEN YT(I,J) = - YT(I,J)
18024 IF FZ = 1 THEN ZT(I,J) = - ZT(I,J)
18030 NEXT J: NEXT I
18035 PRINT
18040 INPUT "NOM NOUVEL OBJET ";O$
18050 GOSUB 3030
18809 IF C = 2 THEN FY = 1
18999 RETURN
```

19000 REM

TRANSLATION OBJET

```
19010 HOME
19020 GOSUB 30000
19040 PRINT "DONNEZ VECTEUR TRANSLATION": PRINT
19050 INPUT "DX=";DX
19060 INPUT "DY=";DY
19070 INPUT "DZ=";DZ
19080 FOR I = 1 TO L
19090 FOR J = 1 TO N(I)
19100 XT(I,J) = XT(I,J) + DX
19110 YT(I,J) = YT(I,J) + DY
19120 ZT(I,J) = ZT(I,J) + DZ
19130 NEXT J: NEXT I
19140 PRINT : PRINT "NOM OBJET TRANSLATE": PRINT
19150 INPUT "";O$
19160 GOSUB 3030
19999 RETURN
20000 REM
```

ROTATION OBJET

```
20010 HOME
20020 GOSUB 30000
20025 PRINT
20030 PRINT "ROTATION AXE PARALLELE A": PRINT
20040 PRINT "1-L'AXE OX : ROULIS": PRINT
20050 PRINT "2-L'AXE OY : TANGAGE": PRINT
20060 PRINT "3-L'AXE OZ : LACET": PRINT
20070 INPUT "VOTRE CHOIX";C
20080 ON C GOSUB 25000,26000,27000
20999 RETURN
21000 REM
```

FUSION OBJETS

```
21010 HOME
21020 PRINT "FIN DE FUSION,TAPEZ LE MOT FIN": PRINT
21030 PRINT "PREMIER OBJET A FUSIONNER": PRINT
21035 GOSUB 30000
21040 INPUT "NOM OBJET A FUSIONNER";O$
21042 IF O$ = "FIN" THEN 21500
21050 PRINT D$"OPEN";O$
21060 PRINT D$"READ";O$
21062 REM
```

OPERATION DE FUSION

```
21065 REM LL=NOMBRE DE CHAINES DU NOUVEL OBJET
21070 INPUT LL
21080 FOR I = L + 1 TO L + LL
21090 INPUT N(I)
21100 NEXT I
21110 FOR I = L + 1 TO L + LL
21120 FOR J = 1 TO N(I)
21130 INPUT XT(I,J): INPUT YT(I,J): INPUT ZT(I,J)
21140 NEXT J: NEXT I
21150 PRINT D$"CLOSE";O$
21155 L = L + LL
21160 GOTO 21040
21500 REM FIN DE FUSION
21510 HOME
21520 PRINT : INPUT "NOM DE L'OBJET GLOBAL ";O$
21530 GOSUB 3030
21999 RETURN
22000 REM
```

HOMOTHETIE

```
22010 HOME
22020 GOSUB 30000
22040 PRINT "COORDONNEES CENTRE HOMOTHETIE": PRIN
T
22050 INPUT "XC=";XC
22060 INPUT "YC=";YC
22070 INPUT "ZC=";ZC
22080 PRINT : INPUT "COEFFICIENT HOMOTHETIE ";HO: PR
INT
22090 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
22100 X1 = XT(I,J) - XC
22110 Y1 = YT(I,J) - YC
22120 Z1 = ZT(I,J) - ZC: REM
```

```

22130 X2 = X1 * HO
22135 Y2 = Y1 * HO
22140 Z2 = Z1 * HO: REM
22145 XT(I,J) = X2 + XC
22150 YT(I,J) = Y2 + YC
22160 ZT(I,J) = Z2 + ZC: REM
22170 NEXT J: NEXT I
22999 RETURN
23000 REM

```

#### AFFINITE

```

23010 HOME :
23015 FX = 0: FY = 0: FZ = 0
23020 GOSUB 30000
23030 PRINT "1- X=0 PLAN YOZ": PRINT
23040 PRINT "2- Y=0 PLAN XOZ": PRINT
23050 PRINT "3- Z=0 PLAN XOY": PRINT
23060 INPUT "VOTRE CHOIX"; C
23100 IF C = 1 THEN FX = 1
23110 IF C = 2 THEN FY = 1
23120 IF C = 3 THEN FZ = 1
23125 PRINT : INPUT "COEF AFFINITE"; AF: PRINT
23130 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
23140 IF FX = 1 THEN XT(I,J) = AF * XT(I,J)
23150 IF FY = 1 THEN YT(I,J) = AF * YT(I,J)
23160 IF FZ = 1 THEN ZT(I,J) = AF * ZT(I,J): REM
23170 NEXT J: NEXT I
23999 RETURN
24000 REM

```

#### HOMOTOPIE

```

24010 FX = 0: FY = 0: FZ = 0
24020 GOSUB 30000
24052 PRINT : INPUT "AMPLITUDE DEFORMATION "; A: PRIN
T
24054 INPUT "LARGEUR DEFORMATION "; B: PRINT
24060 PRINT "1-SELON OX": PRINT
24070 PRINT "2-SELON OY": PRINT
24080 PRINT "3-SELON OZ": PRINT
24090 INPUT "VOTRE CHOIX"; C: PRINT
24092 IF C = 1 THEN FX = 1
24093 IF C = 2 THEN FY = 1
24094 IF C = 3 THEN FZ = 1
24100 FOR I = 1 TO L: FOR J = 1 TO N(I): REM
24110 X1 = XT(I,J): Y1 = YT(I,J): Z1 = ZT(I,J)
24140 R2 = (FX + FY) * Z1 * Z1 + (FX + FZ) * Y1 * Y1 + (FZ + FY
) * X1 * X
1
24150 H = A * (EXP (- (R2 / (2 * B * B))): REM
24160 XT(I,J) = XT(I,J) + FX * H
24170 YT(I,J) = YT(I,J) + FY * H
24180 ZT(I,J) = ZT(I,J) + FZ * H
24220 NEXT J: NEXT I
24999 RETURN
25000 REM

```

#### ROTATION AXE // OX

```

25010 HOME
25020 PRINT "TRACE AXE DANS PLAN YOZ": PRINT
25030 INPUT "YA="; YA: PRINT : INPUT "ZA="; ZA: PRINT
25040 INPUT "ANGLE,DEGRES: "; AR
25050 REM AR ANGLE DE ROULIS
25060 AR = AR * 3.1416 / 180
25070 CR = COS (AR): SR = SIN (AR)
25080 FOR I = 1 TO L: FOR J = 1 TO N(I)
25090 X1 = XT(I,J)
25100 Y1 = YT(I,J) - YA
25110 Z1 = ZT(I,J) - ZA: REM
25120 X2 = X1
25130 Y2 = Y1 * CR - Z1 * SR
25140 Z2 = Y1 * SR + Z1 * CR: REM
25150 XT(I,J) = X2
25160 YT(I,J) = Y2 + YA
25170 ZT(I,J) = Z2 + ZA: REM
25180 NEXT J: NEXT I
25999 RETURN

```

```

26000 REM
ROTATION AXE // OY

26010 HOME
26020 PRINT "TRACE AXE DANS PLAN XOZ": PRINT
26030 INPUT "XA=";XA: PRINT : INPUT "ZA=";ZA: PRINT
26040 INPUT "ANGLE DEGRES";A1
26050 A1 = A1 * 3.1416 / 180
26070 REM A1 ANGLE DE TANGAGE
26075 CT = COS (A1):ST = SIN (A1)
26080 FOR I = 1 TO L: FOR J = 1 TO N(I): REM

26100 X1 = XT(I,J) - XA
26110 Y1 = YT(I,J)
26120 Z1 = ZT(I,J) - ZA: REM

26140 X2 = Z1 * ST + X1 * CT
26150 Y2 = Y1
26160 Z2 = Z1 * CT - X1 * ST: REM
26170 XT(I,J) = X2 + XA
26180 YT(I,J) = Y2
26190 ZT(I,J) = Z2 + ZA: REM

26200 NEXT J: NEXT I
26999 RETURN
27000 REM

```

#### ROTATION AXE // OZ

```

27010 HOME
27020 PRINT "TRACE AXE DANS PLAN XYO": PRINT
27030 INPUT "XA=";XA: PRINT : INPUT "YA=";YA: PRINT
27040 INPUT "ANGLE,DEGRES";AL
27050 AL = AL * 3.1416 / 180
27060 REM AL ANGLE DE LACET
27070 CL = COS (AL):SL = SIN (AL)
27080 FOR I = 1 TO L: FOR J = 1 TO N(I): REM

27090 X1 = XT(I,J) - XA
27100 Y1 = YT(I,J) - YA
27110 Z1 = ZT(I,J): REM

27120 X2 = X1 * CL - Y1 * SL
27130 Y2 = X1 * SL + Y1 * CL
27140 Z2 = Z1: REM

27150 XT(I,J) = X2 + XA
27160 YT(I,J) = Y2 + YA
27170 ZT(I,J) = Z2: REM

27180 NEXT J: NEXT I
27999 RETURN
28000 REM

```

#### CREATION D'UN FICHIER BLOC

```

28010 HOME
28020 INPUT "NOMBRE D'OBJETS ";E
28030 FOR I = 1 TO E
28040 PRINT "OBJET NUMERO ";I
28050 INPUT EL$(I): PRINT
28060 NEXT I
28070 INPUT "NOM DE CE BLOC ";BL$: PRINT
28080 PRINT D$"OPEN";BL$
28090 PRINT D$"WRITE";BL$
28100 PRINT E
28110 FOR I = 1 TO E
28120 PRINT EL$(I)
28130 NEXT I
28140 PRINT D$"CLOSE";BL$
28999 RETURN
29000 REM
LECTURE FICHIER BLOC

```

```

29005 HOME : INPUT "NOM DU BLOC ";BL$: PRINT
29010 D$ = CHR$(13) + CHR$(4)
29020 PRINT D$"OPEN";BL$
29030 PRINT D$"READ";BL$
29040 INPUT E
29050 FOR I = 1 TO E
29060 INPUT EL$(I)
29065 NEXT I
29070 PRINT D$"CLOSE";BL$
29080 FOR I = 1 TO E: PRINT EL$(I): NEXT : FOR TT = 1 TO 2
000: NEXT TT
29999 RETURN

```

```

30000 REM
L'OBJET EST-IL EN MEMOIRE?

30010 PRINT : INPUT "L'OBJET EST-IL EN MEMOIRE ? ";R$:
PRINT
30020 IF LEFT$(R$,1) = "O" THEN 30999
30030 INPUT "NOM DE L'OBJET: ";O$: PRINT
30040 GOSUB 4020
30999 RETURN
40000 REM

CHAIN PANGRAPHE DESSIN

40010 PRINT D$"BLOADCHAIN,A520"
40020 CALL 520"PANGRAPHE DESSIN"

5000 REM

MODIFIER UN OBJET

5010 HOME
5020 GOSUB 30000
5030 PRINT : PRINT "OBJET DE ";L;" CHAINES": PRINT
5040 INPUT "DONNEZ L'INDICE DE CHAINE ";I
5050 INPUT "DONNEZ INDICE POINT SUR LA CHAINE ";J
5060 PRINT
5070 INPUT "XT(I,J)='";XT(I,J): PRINT
5080 INPUT "YT(I,J)='";YT(I,J): PRINT
5090 INPUT "ZT(I,J)='";ZT(I,J): PRINT
5100 INPUT "UN AUTRE POINT ";R$
5110 IF LEFT$(R$,1) = "O" THEN 5040
5120 GOSUB 3020
5998 END

```

# PANGRAPHÉ DESSIN

Cette section représente la deuxième partie de cet ensemble.  
Le Menu de PANGRAPHÉ sera semblable à celui de  
PANGRAPHÉ OBJET :

- 1 - (CRÉER UN OBJET)
- 2 - (STOCKER UN OBJET)
- 3 - CHARGER UN OBJET      PANGRAPHÉ DESSIN
- 4 - (COMPLÉTER UN OBJET)
- 5 - (MODIFIER UN OBJET)
- 6 - PRÉSENTER UN OBJET
- 7 - STOCKER UNE IMAGE
- 8 - CHARGER UNE IMAGE
- 9 - (MANIPULER UN OBJET)
- 10 - (DEFINIR UN BLOC D'OBJETS)
- 11 - CONSULTER UN FICHIER BLOC
- 12 -
- 13 -
- 14 -
- 15 -
- 16 -
- 17 - AFFICHER LE CATALOGUE
- 18 - SORTIR DU PROGRAMME

VOTRE CHOIX ? ■

Comme on peut le voir dans ce menu, les parenthèses sont " inversées ". Elle encadreront les options appartenant à PANGRAPHÉ OBJET.

```

]LOADPANGRAPHÉ DESSIN
JLIST

0 IF FD = 1 THEN 5
1 L = 0: REM PANGRAPHÉ DESSIN 5/2/84
2 DIM XT(30,9),YT(30,9),ZT(30,9),X(30,9),Y(30,9),N(30):FD = 1
3 D$ = CHR$(13) + CHR$(4)
5 TEXT : HOME
6 IF C < > 0 THEN 330
10 HOME
20 REM

PROGRAMME MENU

30 VTAB 15: HTAB 15
40 PRINT "VOULEZ-VOUS:"
50 FOR TE = 1 TO 500: NEXT
60 HOME
70 PRINT
80 VTAB 3: HTAB 25: PRINT "PANGRAPHÉ DESSIN"
90 VTAB 1
100 PRINT "1-(CREER UN OBJET)"
110 PRINT "2-(STOCKER UN OBJET)"
120 PRINT "3-CHARGER UN OBJET"
130 PRINT "4-(COMPLETER UN OBJET)"
140 PRINT "5-(MODIFIER UN OBJET)"
150 PRINT "6-REPRESENTER UN OBJET"
160 PRINT "7-STOCKER UNE IMAGE"
170 PRINT "8-CHARGER UNE IMAGE"
180 PRINT "9-(MANIPULER UN OBJET)"
190 PRINT "10-(DEFINIR UN BLOC D'OBJETS)"
200 PRINT "11-CONSULTER UN FICHIER BLOC"
210 PRINT "12-"
220 PRINT "13-"
230 PRINT "14-"
240 PRINT "15-"
250 PRINT "16-"
260 PRINT "17-AFFICHER LE CATALOGUE"
270 PRINT "18-SORTIR DU PROGRAMME"
280 INPUT "VOTRE CHOIX":C
300 IF C = 17 THEN PRINT D$"CATALOG": GET A$: GET A$
310 IF C = 18 THEN END
320 IF C = 19 THEN PRINT D$"RUNPANGRAPHÉ PRINCIPAL"
330 ON C GOSUB 40000,40000,4000,40000,40000,40000,7000,8000,90
00,40000,40000,2
9000
999 GOTO 10

```

au début, même système de FLAG FD, même test sur la valeur de C.

les options absentes, renvoient à la routine de transfert 40 000, encore présente :

```

40000 REM
CHAINAGE SUR P/OBJET
40010 PRINT D$"BLOADCHAIN,A520"
40020 CALL 520"PANGRAPHÉ OBJET"

```

Pour dessiner un objet, il faudra commencer par le charger en mémoire centrale. Ce qui fait que l'on trouvera encore ici la même routine de chargement :

```
4000 REM
CHARGEMENT OBJET

4010 INPUT "NOM DE L'OBJET ";O$
4020 D$ = CHR$(4)
4030 PRINT D$"OPEN";O$
4040 PRINT D$"READ";O$
4050 INPUT L
4060 FOR I = 1 TO L
4070 INPUT N(I)
4080 NEXT I
4090 FOR I = 1 TO L
4100 FOR J = 1 TO N(I)
4110 INPUT XT(I,J): INPUT YT(I,J): INPUT ZT(I,J)
4120 NEXT J: NEXT I
4130 PRINT D$"CLOSE";O$
4999 RETURN
5000 GOSUB 2000
5999 RETURN
```

Au rayon des choses déjà connues :

```
29000 REM
LECTURE FICHIER BLOC

29005 HOME : INPUT "NOM DU BLOC ";BL$: PRINT
29010 D$ = CHR$(4)
29020 PRINT D$"OPEN";BL$
29030 PRINT D$"READ";BL$
29040 INPUT E
29050 FOR I = 1 TO E
29060 INPUT EL$(I)
29065 NEXT I
29070 PRINT D$"CLOSE";BL$
29080 FOR I = 1 TO E: PRINT EL$(I): NEXT I
29090 FOR TT = 1 TO 2000: NEXT TT
29999 RETURN
```


Tout le reste concerne la fabrication d'images.

# REPRÉSENTER UN OBJET

l'objet (ou les objets) que nous voulons représenter, se situent quelquepart dans l'espace objet  $X_T, Y_T, Z_T$ .


Pour créer une vue en perspective il va falloir indiquer un **POINT D'OBSERVATION** de coordonnées  $X_M, Y_M, Z_M$ .

Il faut ensuite indiquer à la machine dans quelle direction il lui faut "regarder" cet objet, c'est à dire orienter la **FENÊTRE DE VISION**:




Nous savons que nous pouvons à volonté régler l'**OUVERTURE ANGULAIRE AN** de cette fenêtre, c'est à dire notre **FOCALE**, notre **ZOOM**.


L'**ATTITUDE** de l'observateur dépend de trois angles, que l'on retrouve sur un pied d'appareil photographique.




Pour être tout à fait complet, on devrait les faire intervenir tous les trois. Mais, dans cette première version de **PANGRAPHE** nous ne retiendrons que l'**AZIMUT** et le **SITE**. On a en général pas l'habitude de regarder un objet en se tordant le cou de côté. C'est mauvais pour les vertébres, et on risque d'attraper un torticolis.



L'angle de dévers devra être ultérieurement introduit si on souhaite par exemple représenter sur l'écran ce que voit un pilote pendant un virage (simulateur de pilotage



La première idée consisterait à indiquer à l'ordinateur, soi-même, les angles d'azimut et de site. Mais cela impose un calcul pénible, que l'on peut tout aussi bien confier à la machine.




$(\vec{O}X_T, \vec{O}Y_T, \vec{O}Z_T)$  représente le **RÉFÉRENTIEL ABSOLU**, vis à vis duquel on définit les coordonnées  $(X_T, Y_T, Z_T)$  des **POINTS OBJETS**.


M est le point d'observation, de coordonnées  $(X_M, Y_M, Z_M)$  et le vecteur  $MG$  la direction de pointage. On va faire intervenir un trièdre trirectangle  $(\vec{M}X_L, \vec{M}Y_L, \vec{M}Z_L)$  centré en M et dont les axes sont parallèles aux axes du trièdre  $(\vec{O}X_T, \vec{O}Y_T, \vec{O}Z_T)$ .

Sur cette figure, on voit très bien les angles **A** (Azimut) et **S** (site).

La visée se fait en direction d'un point **G**  $(X_G, Y_G, Z_G)$ , et c'est cet ensemble de coordonnées qui sera fourni à la machine. Si le point **G** se trouve être le centre de gravité de l'objet, par exemple, on aura toutes les chances de regarder dans la bonne direction.




A ce stade, on pourrait se dire : pourquoi ne pas inclure ces coordonnées objet XG, YG, ZG dans le fichier OBJET, et stocker cette information. Lors du dessin, le pointage pourrait se faire automatiquement. Mais cela serait trop restrictif et trop rigide. Si nous voulons par exemple représenter une place de village, avec plusieurs objets-maison : le point de visée pourra être le centre de la place.



Quand on tracera l'objet "MAIRIE", la direction de visée ne passera évidemment pas par son centre.

Le problème est surtout de ne pas oublier où se trouve un objet mémorisé sur la disquette. Sinon on risque de perdre pas mal de temps à le rechercher. Une solution consiste à faire figurer les coordonnées de son centre dans le nom de fichier.



Exemple : cet objet s'appellera MAIRIE 8Ø-4Ø-Ø ou mieux : MAIRIE(8Ø/4Ø/Ø)

le vecteur  $\vec{MG}$  a pour composantes



$$\left\{ \begin{array}{l} CX = XG - XM \\ CY = YG - YM \\ CZ = ZG - ZM \end{array} \right.$$

on connaît la suite d'instructions permettant de calcul l'azimut A et le site S. Il faut introduire maintenant un troisième trièdre LIÉ À L'OBSERVATEUR :

L'axe  $\vec{Mu}$  est colinéaire à  $\vec{MG}$ .

L'axe  $\vec{MV}$  est horizontal

L'axe  $\vec{MW}$  complète ce trièdre trirectangle



Si T est un point de l'objet, ses coordonnées "apparentes" par rapport à ce nouveau trièdre, seront  $(XA, YA, ZA)$

à partir de ces coordonnées apparentes nous pourront calculer l'azimut et le site "apparents", qui deviendront les coordonnées écran.

On aura :

$$\left\{ \begin{array}{l} XA = \vec{MT} \cdot \vec{Mu} \\ YA = \vec{MT} \cdot \vec{MV} \\ ZA = \vec{MT} \cdot \vec{MW} \end{array} \right.$$

avec  $\vec{MT} \left\{ \begin{array}{l} XT - XM \\ YT - YM \\ ZT - ZM \end{array} \right.$

$$\xrightarrow{MU} \begin{cases} XU = \cos S \cos A \\ YU = \cos S \sin A \\ ZU = \sin S \end{cases} \quad \xrightarrow{MV} \begin{cases} XV = -\sin A \\ YV = \cos A \\ ZV = 0 \end{cases} \quad \xrightarrow{MW} \begin{cases} XW = -\sin S' \cos A \\ YW = -\sin S' \sin A \\ ZW = \cos S' \end{cases}$$

On posera :  $\begin{cases} \cos A = CT \\ \sin A = ST \\ \cos S = CK \\ \sin S = SK \end{cases}$

d'où

$$\xrightarrow{MU} \begin{cases} XU = CK * CT \\ YU = CK * ST \\ ZU = SK \end{cases} \quad \xrightarrow{MV} \begin{cases} XV = -ST \\ YV = CT \\ ZV = 0 \end{cases} \quad \xrightarrow{MW} \begin{cases} XW = -SK * CT \\ YW = -SK * ST \\ ZW = CK \end{cases}$$

$$\begin{cases} XA = XL * XU + YL * YU + ZL * ZU \\ YA = XL * XV + YL * YV + ZL * ZV \\ ZA = XL * XW + YL * YW + ZL * ZW \end{cases}$$

produits scalaires  
donnant  
les COORDONNÉES  
APPARENTES.

Le programme suivant va matérialiser ces formules.

7000 REM

REPRESENTER UN OBJET

```

7010 TEXT : HOME
7020 PRINT : PRINT
7030 INPUT "OUVERTURE ANGULAIRE "; AN: PRINT
7040 PRINT "POINT D'OBSERVATION"
7050 PRINT : INPUT "X="; XM
7060 INPUT "Y="; YM
7070 INPUT "Z="; ZM
7080 PRINT : PRINT "CENTRE G DE L'IMAGE": PRINT : INPUT
T "XG="; XG: INPUT
"YG="; YG: INPUT "ZG="; ZG
7090 CX = XG - XM: CY = YG - YM: CZ = ZG - ZM
7100 DI = SQR (CX * CX + CY * CY)
7110 IF DI = 0 AND CZ > 0 THEN TE = 0: KI = 3.1416 / 2: GOTO
7210
7120 IF DI = 0 AND CZ < 0 THEN TE = 0: KI = -3.1416 / 2: GO
TO 7210
7130 KI = ATN (CZ / DI)
7140 IF CX = 0 AND CY > 0 THEN TE = 1.57: GOTO 7210
7150 IF CX = 0 AND CY < 0 THEN TE = -1.57: GOTO 7210
7160 TE = ATN (CY / CX)
7170 IF CX < 0 AND CY > 0 THEN TE = 3.1416 + TE
7180 IF CX < 0 AND CY < 0 THEN TE = 3.1416 + TE
7190 IF CX < 0 AND CY = 0 THEN TE = 3.1416
7200 IF CX = 0 AND CY < 0 THEN TE = -3.1416 / 2
7210 AZ = TE * 170 / 3.1416: SI = KI * 170 / 3.1416
7220 REM AZ AZIMUT: SI SITE
7230 CT = COS (TE): ST = SIN (TE)
7240 CK = COS (KI): SK = SIN (KI)
7250 XU = CK * CT: YU = CK * ST: ZU = SK
7260 XV = -ST: YV = CT: ZV = 0
7270 XW = -SK * CT: YW = -SK * ST: ZW = CK
7280 HOME
7290 PRINT "1-OBJET DEJA EN MEMOIRE": PRINT
7300 PRINT "2-DESSIN PAR ELEMENTS": PRINT
7310 PRINT "3-DESSIN PAS A PAS": PRINT
7320 PRINT "4-DESSIN PAR BLOC": PRINT
7330 INPUT "VOTRE CHOIX"; C: PRINT
7340 ON C GOSUB 12000,11000,30000,10000
7998 GET A$: TEXT
7999 RETURN

```

De nouveau, un sous menu et des options

**Option 1**: l'objet est DÉJA en mémoire . Un dessin d'un objet UNIQUE , qu'on aura pas besoin de changer.

**Option 2**: Dessin par éléments .

On indiquera à l'ordinateur que l'on souhaite le tracé d'un décor comprenant un certain nombre d'objets , qui seront saisis par constitution d'un fichier d'éléments EL\$ (i) qui seront ensuite chargés un à un et dessinés

**Option 3**: Dessin pas à pas : dans une alternance de pages graphiques et de pages texte , on pourra surimpressionner , pas à pas , de nouveaux objets

**Option 4**: Dessin par bloc : l'ensemble d'éléments a préalablement été défini et forme un fichier-bloc présent sur la diskette .

le sous programme de trace d'image , à partir des données visuelles et d'un objet présent en mémoire correspondra au sous programme 15000. On va le garder pour la fin .

## OBJET DEJA EN MÉMOIRE :

```
12000 REM
OBJET DEJA EN MÉMOIRE

12010 HGR2 : HCOLOR= 3
12020 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
12030 GOSUB 15000
12040 FOR T = 1 TO 7: PRINT CHR$ (7): NEXT T
12999 RETURN
```

En prime : on trace le cadre ...

## DESSIN PAR ÉLÉMENTS :

```
11000 REM
DESSIN PAR ELEMENTS

11010 HOME
11020 INPUT "NOMBRE D'ÉLÉMENTS ";E
11030 FOR K = 1 TO E
11040 INPUT "NOM ELEMENT ";EL$(K)
11045 NEXT K
11050 HGR2 : HCOLOR= 3
11060 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
11070 FOR K = 1 TO E
11080 O$ = EL$(K)
11090 GOSUB 4020
11100 GOSUB 15000
11110 NEXT K
11120 FOR T = 1 TO 7: PRINT CHR$(7): NEXT T
11999 RETURN
```

Première partie : Série des éléments constituant le fichier EL\$(K)

Deuxième partie : tracé

## DESSIN PAS À PAS

```
30000 REM
DESSIN PAS A PAS

30010 HOME
30020 INPUT "NOM PREMIER ELEMENT ";O$: PRINT
30030 GOSUB 4020
30040 HGR2 : HCOLOR= 3
30050 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
30060 GOSUB 15000: FOR TT = 1 TO 7: PRINT CHR$(7): NEXT
    TT
30070 GET A$: GET A$
```

Première partie : on charge le premier élément. On trace le cadre, puis l'image.

Suite de sept "BIPS" sonores (CHR\$(7)).

Puis pause à l'aide de GET A\$

Une suite de **POKE** permet alors de retourner en mode Texte SANS détruire l'image, qui sera tracée en PAGE II et qui restera intacte.

En mode texte, questionnement :  
UN AUTRE ELEMENT ?

Si c'est non, fin de ce sous-programme, par saut en 30999 RETURN

Si c'est oui, nom de l'élément, chargement, retour en mode graphique par des POKE\$ AD HOC et dessin (GOSUB 15000), et nouvelle pause.

```
30080 POKE - 16303,0: POKE - 16300,0: REM TEXTE
30090 INPUT "UN AUTRE ELEMENT ? ";R$: PRINT
30100 IF LEFT$(R$,1) < > "O" THEN 30999
30110 INPUT "NOM DE L'ELEMENT ";O$: PRINT
30120 GOSUB 4020
30130 POKE - 16304,0: POKE - 16299,0: REM PAGEII
30140 GOSUB 15000: FOR TT = 1 TO 7: PRINT CHR$(7): NEXT
      TT
30150 GET A$: GET A$
30160 GOTO 30080
30999 RETURN
```

## DESSIN PAR BLOC :

```
10000 REM
DESSIN PAR BLOC

10010 INPUT "NOM DU BLOC ";BL$: PRINT
10030 HGR2 : HCOLOR= 3
10040 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
10050 FOR K = 1 TO 8
10060 O$ = EL$(K)
10070 GOSUB 4020
10080 GOSUB 15000
10090 NEXT K
10100 FOR T = 1 TO 7: PRINT CHR$(7): NEXT T
10999 RETURN
```

après avoir saisi le nom du Fichier Bloc, l'ordinateur le charge (GOSUB 29000).

Puis tout se déroule de manière très similaire

On ne sait toujours pas comment on crée un dessin.  
Mais on peut donner les sous programmes qui stockent la page graphique II (le dessin) et l'opération inverse.

```

8000 REM
STOCKER PAGE II

8005 HOME
8010 INPUT "NOM DE L'IMAGE ";IM$
8020 D$ = CHR$(4)
8030 PRINT D$;"BSAVE";IM$;"",A$4000,L$2000"
8999 RETURN
9000 REM

CHARGER SUR PAGE II

9005 HOME
9010 INPUT "NOM DE L'IMAGE? ";IM$
9015 HGR2 : HCOLOR= 3
9020 D$ = CHR$(4)
9030 PRINT D$;"BLOAD";IM$;"",A$4000"
9060 GET A$
9999 TEXT : HOME : RETURN

```

Reste le calcul et trace de l' image

# IMAGE

On commence par calculer les coordonnées tridimensionnelles **APPARENTES**, c'est à dire les coordonnées cartésiennes des points dans un référentiel lié à l' observateur .

```

15000 REM
CALCUL ET TRACE IMAGE

15010 FOR I = 1 TO L: FOR J = 1 TO N(I)
15040 XL = XT(I,J) - XM
15050 YL = YT(I,J) - YM
15060 ZL = ZT(I,J) - ZM
15090 XA = XL * XU + YL * YU + ZL * ZU
15100 YA = XL * XV + YL * YV + ZL * ZV
15110 ZA = XL * XW + YL * YW + ZL * ZW

```

Comment convertir ces coordonnées tridimensionnelles en coordonnées de points sur l' écran.


On peut utiliser deux systèmes de repérage de visée.  
le système de l' artilleur : azimuth + siège.  
ou la visée **POLAIRE**.

le repérage de l'artilleur est **BESTIAL**. Il suffit de se reporter aux pages 12 et suivantes



J'ai retrouvé cet objet très ancien, avec lequel les hommes ont pu comprendre les lois de la **PERSPECTIVE**, grâce à lui ils pouvaient calculer ces coordonnées-écran ( $X, Y$ )

- Mais j'ai trouvé aussi cet objectif **FISH-EYE** :  
œil de poisson.



Son **OUVERTURE** atteint 90°.

- Tu vois, j'ai l'impression que les hommes ont fait, dès le départ, une erreur fondamentale sur le mécanisme de la vision. La **VISION** est un **GESTE**. En effet un œil humain immobilisé est pratiquement aveugle. Pour plusieurs raisons. La première est que les cellules sensibles de la rétine sont principalement massées au voisinage de la tache fovéale, de l'axe optique. Et la seconde que ces cellules réagissent non à l'intensité lumineuse, mais à la **VARIATION** de cette intensité. La perception visuelle n'est donc pas statique. L'axe optique humain est en perpétuelle agitation.






SCHÉMA DE  
LA RÉTINE



mouvement de l'axe optique.

- En somme, on voit mal le lien entre ce que les hommes montrent par exemple dans leurs films, et les images peuplant leur rétine ...

Un écran de télévision a une certaine rémanence. lorsque la caméra filme un ensemble de lampes donnant un éclairage très vif, et que cette caméra bouge, ces points lumineux laissent des traînées visibles sur l'écran fluorescent.



TRACE DU MOUVEMENT  
DE LAMPES SUR UN  
ÉCRAN, DU A LA  
RÉMANENCE

du CHAMP VISUEL.

des points d'intérêt.

- En somme, l'image ci-contre donne une certaine idée de l'information dont dispose le cerveau humain pour reconstituer son environnement visuel.

- La VISION est une EXPLORATION permanente : l'axe optique est sans cesse dirigé vers

des points d'intérêt.

- L'œil d'Arthur est équipé pour réaliser cette sorte de POURSUITE de l'objet.

- Mais je ne suis pas sûr que ces angles : AZIMUT, SITE, traduisent la réalité mentale qui était celle des humains. Le pointage évoque plus le canon que l'œil.




- Pourtant les hommes se basent sur ces déplacements en azimut et site, pour déplacer ces canons. Regarde cet antique système de DCA.

- Oui, mais regarde le VISEUR !


- Ah, tu as raison. Il

méritait une POURSUITE en COORDONNÉES POLAIRES.


- Il n'y a pas de viseur à mailles carées ou rectangulaires.




— Ainsi les hommes ont construit des systèmes de **POINTAGE** basés sur le système (azimut, site) pour des raisons de commodités mécaniques. Mais, quand il s'agit de **VISÉE**, ils retrouvent le système qui leur est le plus **NATUREL**, c'est à dire un système de coordonnées **POLAIRES**.



Dans l'opération **DESSIN**, l'écart angulaire vis à vis de l'axe visuel est converti en longueur **P**.



De même, précédemment, l'azimut et le site étaient convertis en **COORDONNÉES ÉCRAN X,Y**




— Tu sais quoi ? On devrait modifier le programme de construction de l'image, en la définissant sur la base de ces **COORDONNÉES POLAIRES**.

— Oui, cela serait plus naturel ....

— Et cela permet d'introduire des **OUVERTURES ANGULAIRES** plus importantes.

— Dans cette **FENÊTRE DE VISÉE**, moi je fais de la claustrophobie ...

# ViSÉE POLAIRE





La croix est la trace de l'axe visuel.

$\alpha$  est l'angle polaire

$\beta$  est l'écart angulaire, vis à vis de la direction axial

Un tel système permet de rendre compte de visées dépassant les capacités des yeux humains.


## LE FISH-EYE



L'écart angulaire peut alors atteindre 90°


Mais on peut pousser plus loin encore.

# L'ŒIL DE MOUCHE




La déviation angulaire  $\beta$  peut alors atteindre la valeur maximale  $180^\circ$ .

Dessiner sera donner une représentation plane, en coordonnées polaires de ces images. Le champ visuel est alors représenté selon un cercle. Et le rayon de ce cercle correspond à la valeur maximale de l'écart angulaire  $\beta$ . Sur l'écran du moniteur nous aurons :




Dans le cas du FISH-EYE le cercle limite correspond à  $\beta = 90^\circ$ .




Un observateur équipé d'un FISH-EYE observe une mire carrée, maillée, dont le plan est perpendiculaire à son axe visuel.

la représentation plane est donnée sur la page suivante.




Passons maintenant en vision totale (œil de la mouche.)  
La représentation plane de cet objet serait :




Il s'est tassé au centre  
de l'écran, puisque  
les valeurs de  $\beta$  de ses  
points sont relativement  
faibles vis à vis de la  
valeur maximale  $\beta = 180^\circ$

Nous découvrons ici un aspect déconcertant de cette représentation.  
Le point qui est situé "derrière la tête de l'observateur"  
se matérialise selon un cercle ( $\beta = 180^\circ$ )  
Convenons d'appeler le point à l'infini dans l'axe de visée  
**le POINT FRONTAL**, et le point situé dans la direction  
diamétralement **le POINT OCCIPITAL**.  
l'image plane du point occipital est donc un cercle.




Si on limite l'écart angulaire maximum à une vingtaine ou à une trentaine de degrés on retrouve une vision très proche de la vision humaine.


On va donc réécrire le programme **PANGRAPHE** de telle manière qu'il puisse intégrer tous ces types de vision, de celle de l'homme à celle de la mouche, en passant par le fish-eye.



- L'Apple II trace les segments à l'aide de l'instruction  
`HPLOT XA,YA TO XB,YB`



- c'est à dire qu'il opère sur des coordonnées-écran cartésiennes, qui il faudra calculer. Mais, dans un mode de tracé **VECTORIEL**, on pourrait utiliser directement des données polaires.
- Quand les humains peuplaient encore la Terre, certains savaient dessiner, et d'autres en étaient parfaitement incapables. Pourquoi ?
- A mon avis, il devait s'agir d'un mauvais accord entre leur hard et leur soft. Le dessin manuel suggère une exploitation de données en coordonnées polaires. Le dessin est un geste. La main se déplace dans la direction polaire, et l'amplitude du geste traduit l'écart angulaire.
- Autrement dit, si un type code dans sa tête en coordonnées cartésiennes, il lui est impossible de dessiner ...
- C'est un problème d'interfaçage.
- Sans doute.



la séquence suivante (qui lève aussi les ambiguïtés sur les Arctangentes) matérialise ce calcul, cette projection sur la sphère d'observation - écran, par visée-polaire.

```

15121 IF XA = 0 AND YA = 0 AND ZA = 0 THEN ZA = .001
15122 RO = SQR (YA * YA + ZA * ZA)
15125 IF XA = 0 THEN B = 90: GOTO 15127
15126 B = ATN (RO / XA):B = B * 180 / 3.1416
15127 IF XA < 0 THEN B = B + 180
15130 IF YA = 0 AND ZA = 0 THEN A = 90: GOTO 15180
15140 IF ZA = 0 AND YA > 0 THEN A = - 90: GOTO 15180
15150 IF ZA = 0 AND YA < 0 THEN A = 90: GOTO 15180
15160 A = - ATN (YA / ZA):A = A * 180 / 3.1416
15170 IF ZA < 0 AND YA > 0 THEN A = A - 180
15172 IF ZA < 0 AND YA < 0 THEN A = A + 180
15174 IF ZA < 0 AND YA = 0 THEN A = 180
15180 X = B * SIN (A * 3.1416 / 180):Y = B * COS (A * 3.1416
/ 180)
15240 X(I,J) = 140 + X * 85 / AN
15250 Y(I,J) = 85 - X * 85 / AN
15260 NEXT J: NEXT I

```

$X(I,J)$  et  $Y(I,J)$  sont les coordonnées - écran.

on va ensuite matérialiser les segments "AB" successifs constituant la chaîne à dessiner. Ils devront être analysés un à un (problème du FENÊTRAGE)

```

15280 FOR I = 1 TO L
15290 FOR J = 1 TO N(I) - 1
15300 XA = X(I,J):YA = Y(I,J):FP = 0:H = 0
15310 XB = X(I,J + 1):YB = Y(I,J + 1)

```

la séquence ci après représente le travail de fenêtrage. On essaie en effet de tracer le segment simple. ment en faisant

HPLOT XA,YA TO XB,YB

message qui ne sera exécuté que si les points A et B sont tous les deux à l'intérieur du cadre. Dans le cas contraire on récupérera un message d'erreur.

# FENÊTRAGE

```


15320 IF YA = YB AND YA > = 0 AND YA < = 190 THEN XL(1)
> = 0:XL(2) = 27
         9:YL(1) = YA:YL(2) = YA:FP = 1: GOTO 15480
15325 IF XA = XB AND XA < 0 THEN 15590
15326 IF XA = XB AND XA > 279 THEN 15590
15330 IF XA = XB AND XA > = 0 AND XA < = 279 THEN XL(1)
= XA:XL(2) = X
         A:YL(1) = 0:YL(2) = 190:FP = 1: GOTO 15480
15335 IF YB = YA AND YA < 0 THEN 15590
15336 IF YB = YA AND YA > 190 THEN 15590
15340 P = (YB - YA) / (XB - XA)
15360 YI = YA - P * XA
15370 IF YI > = 0 AND YI < = 190 THEN FP = 1:H = H + 1:XL(
H) = 0:YL(H)
         = YI
15390 YI = YA + (279 - XA) * P
15400 IF YI > = 0 AND YI < = 190 THEN FP = 1:H = H + 1:XL(
H) = 279:YL(
         H) = YI
15420 XI = XA - YA / P
15430 IF XI > = 0 AND XI < = 279 THEN FP = 1:H = H + 1:XL(H)
> = XI:YL(H)
         ) = 0
15450 XI = XA + (190 - YA) / P
15460 IF XI > = 0 AND XI < = 279 THEN FP = 1:H = H + 1:XL(H)
> = XI:YL(H)
         ) = 190
15470 IF FP = 0 THEN 15590
15480 SA = (XA - XL(1)) * (XA - XL(2)) + (YA - YL(1)) * (YA - YL(
2))
15490 SB = (XB - XL(1)) * (XB - XL(2)) + (YB - YL(1)) * (YB - YL(
2))
15492 S1 = (XL(1) - XA) * (XL(1) - XB) + (YL(1) - YA) * (YL(1) - Y
B)
15494 S2 = (XL(2) - XA) * (XL(2) - XB) + (YL(2) - YA) * (YL(2) - Y
B)
15500 IF SA < = 0 AND SB < = 0 THEN HPLOT XA,YA TO XB
,YB: GOTO 15590
15510 IF SA > 0 AND SB > 0 AND S1 < 0 AND S2 < 0 THEN HP
LOT XL(1),YL(1)
         TO XL(2),YL(2): GOTO 15590
15520 IF SA < = 0 THEN XD = XA:YD = YA
15530 IF SB < = 0 THEN XD = XB:YD = YB
15560 IF S1 < = 0 THEN XF = XL(1):YF = YL(1)
15570 IF S2 < = 0 THEN XF = XL(2):YF = YL(2)
15580 HPLOT XD,YD TO XF,YF
15590 NEXT J: NEXT I
15999 RETURN

```

Dans les ordinateurs actuels il n'est plus nécessaire de programmer le fenêtrage : il est automatique.  
... vestige du passé ...

on étudiera alors les positions relatives des points A et B par rapport aux deux points d'intersection M1 et M2 de la droite AB avec le cadre.

Redressons l'axe OY de l'écran (sur l'Apple II il a "la tête en bas") -



La pente de la droite AB est  $P = (Y_B - Y_A) / (X_B - X_A)$

L'équation de la droite AB est :  $\frac{Y - Y_A}{X - X_A} = \frac{Y_B - Y_A}{X_B - X_A} = P$

Premier élément du cadre  $X = \phi \Rightarrow Y_I = Y_A - P * X_A \quad (\alpha)$


La droite AB ne peut couper le cadre qu'en deux points au plus. Le point sera à retenir si :  $\phi < Y_I < 19\phi$

Second élément du cadre :  $X = 27\phi \Rightarrow Y_I = Y_A + (27\phi - X_A) * P \quad (\beta)$   
à retenir si  $\phi < Y_I < 19\phi$

Troisième élément :  $Y = \phi \Rightarrow X_I = X_A - Y_A / P \quad (\gamma)$   
à retenir si  $\phi < X_I < 27\phi$

Quatrième élément  $Y = 19\phi \Rightarrow X_I = X_A + (19\phi - Y_A) / P \quad (\delta)$   
à retenir si  $\phi < X_I < 27\phi$ .


Il reste enfin possible que la droite AB ne coupe pas le cadre :




Comme on le verra plus loin, une petite astuce de programmation permet de déterminer en une seule fois les éventuels points d'intersection M<sub>1</sub> et M<sub>2</sub> de la droite A B avec le cadre. Il restera à déterminer les positions respectives des points A, B, M<sub>1</sub>, M<sub>2</sub> sur la droite.

On se basera pour se faire sur le signe des produits scalaires :


$$\left\{ \begin{array}{l} S_A = \overrightarrow{M_1A} \cdot \overrightarrow{M_2A} \\ S_B = \overrightarrow{M_1B} \cdot \overrightarrow{M_2B} \\ S_1 = \overrightarrow{AM_1} \cdot \overrightarrow{BM_1} \\ S_2 = \overrightarrow{AM_2} \cdot \overrightarrow{BM_2} \end{array} \right.$$




$$\begin{aligned} S_A &< 0 \\ S_B &< 0 \\ S_1 &> 0 \\ S_2 &> 0 \end{aligned}$$



$$\begin{aligned} S_A &> 0 \\ S_B &> 0 \\ S_1 &< 0 \\ S_2 &< 0 \end{aligned}$$



$$\begin{aligned} S_A &< 0 \\ S_B &> 0 \\ S_1 &> 0 \\ S_2 &< 0 \end{aligned}$$



$$\begin{aligned} S_A &> 0 \\ S_B &> 0 \\ S_1 &< 0 \\ S_2 &< 0 \end{aligned}$$


On voit comment l'examen des produits scalaires permet de choisir les extrémités du segment à tracer.

15315

FENETRAGE




$\left\{ \begin{array}{l} 1532 \phi \text{ IF } YA = YB \text{ AND } YA > \phi \text{ AND } YA <= 19 \phi \text{ THEN} \\ XL(1) = \phi : XL(2) = 279 : YL(1) = YA : YL(2) = YA : \\ FP = 1 : GOTU 1548 \phi \end{array} \right.$



$\left\{ \begin{array}{l} 15325 \text{ IF } XA = XB \text{ AND } XA < \phi \text{ THEN } 15590 \end{array} \right.$



$\left\{ \begin{array}{l} 15326 \text{ IF } XA = XB \text{ AND } XA > 279 \text{ THEN } 1559 \phi \end{array} \right.$



$\left\{ \begin{array}{l} 1533 \phi \text{ IF } XA = XB \text{ AND } XA > \phi \text{ AND } YA <= 279 \\ \text{THEN } XL(1) = XA : XL(2) = XA : YL(1) = \phi : YL(2) \\ = 19 \phi : FP = 1 : GOTU 1548 \phi \end{array} \right.$




$\left\{ \begin{array}{l} 15335 \text{ IF } YB = YA \text{ AND } YA < \phi \text{ THEN } 1559 \phi \end{array} \right.$



$\left\{ \begin{array}{l} 15336 \text{ IF } YB = YA \text{ AND } YA > 19 \phi \text{ THEN } 1559 \phi \end{array} \right.$

Calcul de la pente  $1534 \phi P = (YB - YA) / (XB - XA)$


Intersection avec la droite  $X = \phi$   $1536 \phi YI = YA - P * XA$



$1537 \phi \text{ IF } YI >= \phi \text{ AND } YI <= 19 \phi \text{ THEN } FP = 1 : \\ H = H + 1 : XL(H) = \phi : YL(H) = YI$

le bord gauche est coupé par la droite AB?

Intersection avec la droite  $X = 279$



$1539 \phi YI = YA + (279 - XA) * P$


$1540 \phi \text{ IF } YI >= \phi \text{ AND } YI <= 19 \phi \text{ THEN } FP = 1 : \\ H = H + 1 : XL(H) = 279 : YL(H) = YI$

Test pour voir si le bord droit est coupé par AB

Intersection avec la 1542 $\phi$   $X_I = X_A - Y_A / P$   
droite  $Y = \phi$

Test d'intersection 1543 $\phi$  IF  $X_I >= \phi$  AND  $X_I <= 279$   
avec le bord situé THEN FP=1:  
dans  $Y = \phi$

$$H = H + 1 : X_L(H) = X_I : Y_L(H) = \phi$$



Intersection avec 1545 $\phi$   $X_I = X_A + (19\phi - Y_A) / P$   
la droite  $Y = 19\phi$

Test 1546 $\phi$  IF  $X_I >= \phi$  AND  $X_I <= 279$  THEN FP=1:  
1546 $\phi$  THEN FP=1:  
 $H = H + 1 : X_L(H) = X_I : Y_L(H) = 19\phi$

Cas où AB ne coupe pas le cadre:




(Flag reste nul)

Produit Scalaire  $1548\phi$   $SA = (X_A - X_L(1)) * (X_A - X_L(2))$   
 $\vec{M_1A} \cdot \vec{M_2A}$   $+ (Y_A - Y_L(1)) * (Y_A - Y_L(2))$

Produit Scalaire  $1549\phi$   $SB = (X_B - X_L(1)) * (X_B - X_L(2))$   
 $\vec{M_1B} \cdot \vec{M_2B}$   $+ (Y_B - Y_L(1)) * (Y_B - Y_L(2))$


Produit Scalaire  $15492\phi$   $S1 = (X_L(1) - X_A) * (X_L(1) - X_B)$   
 $\vec{AM_1} \cdot \vec{BM_1}$   $+ (Y_L(1) - Y_A) * (Y_L(1) - Y_B)$

Produit Scalaire  $15494\phi$   $S2 = (X_L(2) - X_A) * (X_L(2) - X_B)$   
 $\vec{AM_2} \cdot \vec{BM_2}$   $+ (Y_L(2) - Y_A) * (Y_L(2) - Y_B)$




A et B à l'intérieur  
du segment M1M2

1550 $\phi$  IF  $SA <= \phi$  AND  $SB <= \phi$  THEN  
HPLOT XA,YA TO XB,YB : GOTO 15590



M1 et M2 sont à l'intérieur de AB



1551 $\phi$  IF SA >  $\phi$  AND SB >  $\phi$  AND S1 <  $\phi$   
AND S2 <  $\phi$  THEN HPLT XL(1), YL(1) TO  
XL(2), YL(2) : GOTO 1559 $\phi$

1552 $\phi$  IF SA <=  $\phi$  THEN XD = XA : YD = YA  
(début du segment à tracer)

1553 $\phi$  IF SB <=  $\phi$  THEN XD = XB : YD = YB

1554 $\phi$  IF S1 <=  $\phi$  THEN XF = XL(1) : YF = YL(1)

1555 $\phi$  IF S2 <=  $\phi$  THEN XF = XL(2) : YF = YL(2)


1558 $\phi$  HPLT XD, YD TO XF, YF

1559 $\phi$  NEXT J : NEXT I

15999 RETURN

Bon, maintenant nous avons déjà des éléments pour commencer à dessiner

Crois-tu, Zaks, que nous allons avoir des émotions artistiques ?



# PANGRAPHE DESSIN

```
0 IF FD = 1 THEN 5
1 L = 0: REM PANGRAPHE DESSIN 5/2/84
2 DIM XT(30,9),YT(30,9),ZT(30,9),X(30,9),Y(30,9),N(30):FD = 1
3 D$ = CHR$(13) + CHR$(4)
5 TEXT : HOME
6 IF C < > 0 THEN 330
10 HOME
20 REM
```

## PROGRAMME MENU

```
30 VTAB 15: HTAB 15
40 PRINT "VOULEZ-VOUS:"
50 FOR TE = 1 TO 500: NEXT
60 HOME
70 PRINT
80 VTAB 8: HTAB 25: PRINT "PANGRAPHE DESSIN"
90 VTAB 1
100 PRINT "1-(CREER UN OBJET)"
110 PRINT "2-(STOCKER UN OBJET)"
120 PRINT "3-CHARGER UN OBJET"
130 PRINT "4-(COMPLETER UN OBJET)"
140 PRINT "5-(MODIFIER UN OBJET)"
150 PRINT "6-REPRESENTER UN OBJET"
160 PRINT "7-STOCKER UNE IMAGE"
170 PRINT "8-CHARGER UNE IMAGE"
180 PRINT "9-(MANIPULER UN OBJET)"
190 PRINT "10-(DEFINIR UN BLOC D'OBJETS)"
200 PRINT "11-CONSULTER UN FICHIER BLOC"
210 PRINT "12-"
220 PRINT "13-"
230 PRINT "14-"
240 PRINT "15-"
250 PRINT "16-"
260 PRINT "17-AFFICHER LE CATALOGUE"
270 PRINT "18-SORTIR DU PROGRAMME"
290 INPUT "VOTRE CHOIX ";C
300 IF C = 17 THEN PRINT D$"CATALOG": GET A$: GET A$
310 IF C = 18 THEN END
320 IF C = 19 THEN PRINT D$"RUNPANGRAPHE PRINCIPAL
"
330 ON C GOSUB 40000,40000,4000,40000,40000,7000,8000,90
00,40000,40000,2
9000
999 GOTO 10
4000 REM
```

## CHARGEMENT OBJET

```
4010 INPUT "NOM DE L'OBJET ";O$
4020 D$ = CHR$(4)
4030 PRINT D$"OPEN";O$
4040 PRINT D$"READ";O$
4050 INPUT L
4060 FOR I = 1 TO L
4070 INPUT N(I)
4080 NEXT I
4090 FOR I = 1 TO L
4100 FOR J = 1 TO N(I)
4110 INPUT XT(I,J): INPUT YT(I,J): INPUT ZT(I,J)
4120 NEXT J: NEXT I
4130 PRINT D$"CLOSE";O$
4999 RETURN
5000 GOSUB 2000
5999 RETURN
```

7000 REM

REPRESENTER UN OBJET

```
7010 TEXT : HOME
7020 PRINT : PRINT
7030 INPUT "OUVERTURE ANGULAIRE ";AN: PRINT
7040 PRINT "POINT D'OBSERVATION"
7050 PRINT : INPUT "X=";XM
7060 INPUT "Y=";YM
7070 INPUT "Z=";ZM
7080 PRINT : PRINT "CENTRE G DE L'IMAGE": PRINT : INPUT
T "XG=";XG: INPUT
"YG=";YG: INPUT "ZG=";ZG
7090 CX = XG - XM:CY = YG - YM:CZ = ZG - ZM
7100 DI = SQR (CX * CX + CY * CY)
7110 IF DI = 0 AND CZ > 0 THEN TE = 0:KI = 3.1416 / 2: GOTO
7210
7120 IF DI = 0 AND CZ < 0 THEN TE = 0:KI = - 3.1416 / 2: GO
TO 7210
7130 KI = ATN (CZ / DI)
7140 IF CX = 0 AND CY > 0 THEN TE = 1.57: GOTO 7210
7150 IF CX = 0 AND CY < 0 THEN TE = - 1.57: GOTO 7210
7160 TE = ATN (CY / CX)
7170 IF CX < 0 AND CY > 0 THEN TE = 3.1416 + TE
7180 IF CX < 0 AND CY < 0 THEN TE = 3.1416 + TE
7190 IF CX < 0 AND CY = 0 THEN TE = 3.1416
7200 IF CX = 0 AND CY < 0 THEN TE = - 3.1416 / 2
7210 AZ = TE * 170 / 3.1416:SI = KI * 170 / 3.1416
7220 REM AZ AZIMUT:SI SITE
7230 CT = COS (TE):ST = SIN (TE)
7240 CK = COS (KI):SK = SIN (KI)
7250 XU = CK * CT:YU = CK * ST:ZU = SK
7260 XV = - ST:YV = CT:ZV = 0
7270 XW = - SK * CT:YW = - SK * ST:ZW = CK
7280 HOME
7290 PRINT "1-OBJET DEJA EN MEMOIRE": PRINT
7300 PRINT "2-DESSIN PAR ELEMENTS": PRINT
7310 PRINT "3-DESSIN PAS A PAS": PRINT
7320 PRINT "4-DESSIN PAR BLOC": PRINT
7330 INPUT "VOTRE CHOIX ";C: PRINT
7340 ON C GOSUB 12000,11000,30000,10000
7998 GET A$: TEXT
7999 RETURN
8000 REM
```

STOCKER PAGE II

```
8005 HOME
8010 INPUT "NOM DE L'IMAGE ";IM$
8020 D$ = CHR$(4)
8030 PRINT D$"BSAVE";IM$;"A$4000,L$2000"
8999 RETURN
9000 REM
```

CHARGER SUR PAGE II

```
9005 HOME
9010 INPUT "NOM DE L'IMAGE? ";IM$
9015 HGR2 : HCOLOR= 3
9020 D$ = CHR$(4)
9030 PRINT D$"BLOAD";IM$;"A$4000"
9060 GET A$
9999 TEXT : HOME : RETURN
10000 REM
```

DESSIN PAR BLOC

```
10010 INPUT "NOM DU BLOC ";BL$: PRINT
10030 HGR2 : HCOLOR= 3
10040 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
10050 FOR K = 1 TO E
10060 O$ = EL$(K)
10070 GOSUB 4020
10080 GOSUB 15000
10090 NEXT K
10100 FOR T = 1 TO 7: PRINT CHR$(7): NEXT T
10999 RETURN
```

```

11000 REM
DESSIN PAR ELEMENTS

11010 HOME
11020 INPUT "NOMBRE D'ELEMENTS ";E
11030 FOR K = 1 TO E
11040 INPUT "NOM ELEMENT ";EL$(K)
11045 NEXT K
11050 HGR2 : HCOLOR= 3
11060 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
11070 FOR K = 1 TO E
11080 O$ = EL$(K)
11090 GOSUB 4020
11100 GOSUB 15000
11110 NEXT K
11120 FOR T = 1 TO 7: PRINT CHR$(7): NEXT T
11999 RETURN
12000 REM

```

#### OBJET DEJA EN MEMOIRE

```

12010 HGR2 : HCOLOR= 3
12020 HPLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
12030 GOSUB 15000
12040 FOR T = 1 TO 7: PRINT CHR$(7): NEXT T
12999 RETURN
15000 REM

```

#### CALCUL ET TRACE IMAGE

```

15010 FOR I = 1 TO L: FOR J = 1 TO N(I)
15040 XL = XT(I,J) - XM
15050 YL = YT(I,J) - YM
15060 ZL = ZT(I,J) - ZM
15090 XA = XL * XU + YL * YU + ZL * ZU
15100 YA = XL * XV + YL * YV + ZL * ZV
15110 ZA = XL * XW + YL * YW + ZL * ZW
15121 IF XA = 0 AND YA = 0 AND ZA = 0 THEN ZA = .001
15122 RO = SQR (YA * YA + ZA * ZA)
15125 IF XA = 0 THEN B = 90: GOTO 15127
15126 B = ATN (RO / XA):B = B * 180 / 3.1416
15127 IF XA < 0 THEN B = B + 180
15130 IF YA = 0 AND ZA = 0 THEN A = 90: GOTO 15180
15140 IF ZA = 0 AND YA > 0 THEN A = -90: GOTO 15180
15150 IF ZA = 0 AND YA < 0 THEN A = 90: GOTO 15180
15160 A = -ATN (YA / ZA):A = A * 180 / 3.1416
15170 IF ZA < 0 AND YA > 0 THEN A = A - 180
15172 IF ZA < 0 AND YA < 0 THEN A = A + 180
15174 IF ZA < 0 AND YA = 0 THEN A = 180
15180 X = B * SIN (A * 3.1416 / 180):Y = B * COS (A * 3.1416 / 180)
15240 X(I,J) = 140 + X * 85 / AN
15250 Y(I,J) = 85 - 85 * Y / AN
15260 NEXT J: NEXT I
15280 FOR I = 1 TO L
15290 FOR J = 1 TO N(I) - 1
15300 XA = X(I,J):YA = Y(I,J):FP = 0:H = 0
15310 XB = X(I,J + 1):YB = Y(I,J + 1)
15320 IF YA = YB AND YA > = 0 AND YA < = 190 THEN XL(1) = 0:XL(2) = 27
9:YL(1) = YA:YL(2) = YA:FP = 1: GOTO 15480
15325 IF XA = XB AND XA < 0 THEN 15590
15326 IF XA = XB AND XA > 279 THEN 15590
15330 IF XA = XB AND XA > = 0 AND XA < = 279 THEN XL(1) = XA:XL(2) = X
A:YL(1) = 0:YL(2) = 190:FP = 1: GOTO 15480
15335 IF YB = YA AND YA < 0 THEN 15590
15336 IF YB = YA AND YA > 190 THEN 15590
15340 P = (YB - YA) / (XB - XA)
15360 YI = YA - P * XA
15370 IF YI > = 0 AND YI < = 190 THEN FP = 1:H = H + 1:XL(H) = 0:YL(H) = YI
15390 YI = YA + (279 - XA) * P
15400 IF YI > = 0 AND YI < = 190 THEN FP = 1:H = H + 1:XL(H) = 279:YL(H) = YI
15420 XI = XA - YA / P
15430 IF XI > = 0 AND XI < = 279 THEN FP = 1:H = H + 1:XL(H) = XI:YL(H) = 0
15450 XI = XA + (190 - YA) / P
15460 IF XI > = 0 AND XI < = 279 THEN FP = 1:H = H + 1:XL(H) = XI:YL(H) = 190
15470 IF FP = 0 THEN 15590
15480 SA = (XA - XL(1)) * (XA - XL(2)) + (YA - YL(1)) * (YA - YL(2))

```

```

15490 SB = (XB - XL(1)) * (XB - XL(2)) + (YB - YL(1)) * (YB - YL
(2))
15492 S1 = (XL(1) - XA) * (XL(1) - XB) + (YL(1) - YA) * (YL(1) - Y
B)
15494 S2 = (XL(2) - XA) * (XL(2) - XB) + (YL(2) - YA) * (YL(2) - Y
B)
15500 IF SA < = 0 AND SB < = 0 THEN H PLOT XA,YA TO XB
,YB: GOTO 15590
15510 IF SA > 0 AND SB > 0 AND S1 < 0 AND S2 < 0 THEN HP
LOT XL(1),YL(1)
TO XL(2),YL(2): GOTO 15590
15520 IF SA < = 0 THEN XD = XA: YD = YA
15530 IF SB < = 0 THEN XD = XB: YD = YB
15560 IF S1 < = 0 THEN XF = XL(1): YF = YL(1)
15570 IF S2 < = 0 THEN XF = XL(2): YF = YL(2)
15580 H PLOT XD,YD TO XF,YF
15590 NEXT J: NEXT I
15999 RETURN
29000 REM
LECTURE FICHIER BLOC

```

```

29005 HOME : INPUT "NOM DU BLOC ";BL$: PRINT
29010 D$ = CHR$(4)
29020 PRINT D$"OPEN";BL$
29030 PRINT D$"READ";BL$
29040 INPUT E
29050 FOR I = 1 TO E
29060 INPUT EL$(I)
29065 NEXT I
29070 PRINT D$"CLOSE";BL$
29080 FOR I = 1 TO E: PRINT EL$(I): NEXT I
29090 FOR TT = 1 TO 2000: NEXT TT
29999 RETURN
30000 REM

```

DESSIN PAS A PAS

```

30010 HOME
30020 INPUT "NOM PREMIER ELEMENT ";O$: PRINT
30030 GOSUB 4020
30040 HGR2 : HCOLOR= 3
30050 H PLOT 0,0 TO 279,0 TO 279,190 TO 0,190 TO 0,0
30060 GOSUB 15000: FOR TT = 1 TO 7: PRINT CHR$(7): NEXT
TT
30070 GET A$: GET A$
30080 POKE - 16303,0: POKE - 16300,0: REM TEXTE
30090 INPUT "UN AUTRE ELEMENT ? ";R$: PRINT
30100 IF LEFT$(R$,1) < > "O" THEN 30999
30110 INPUT "NOM DE L'ELEMENT ";O$: PRINT
30120 GOSUB 4020
30130 POKE - 16304,0: POKE - 16299,0: REM PAGEII
30140 GOSUB 15000: FOR TT = 1 TO 7: PRINT CHR$(7): NEXT
TT
30150 GET A$: GET A$
30160 GOTO 30080
30999 RETURN
40000 REM

```

CHAINAGE SUR P/OBJET

```

40010 PRINT D$"BLOADCHAIN,A520"
40020 CALL 520"PANGRAPHE OBJET"

```


### Remarque :

l'ensemble **PANGRAPHE-OBJET + PANGRAPHE DESSIN** a été conçu pour pouvoir être aisément récupéré et modifié par l'utilisateur, qui trouvera là une initiation à la **CONCEPTION ASSISTEE SUR ORDINATEUR (CAO)**. Les Menus comportent des "lignes vides", que le lecteur pourra remplir au gré de son imagination.

1999 : Le **BASIC** moderne ne comportent plus de "lignes programmes". Les têtes de sous-programmes sont repérées à l'aide "d'étiquettes", ce qui donne encore plus de souplesse à l'ensemble. Toujours est-il que l'ensemble proposé est **MODULAIRE** et que l'utilisateur peut ajouter des sous-programmes à l'infini pour enrichir celui-ci.

Si vous dotez **PANGRAPHE DESSIN** de quelques lignes supplémentaires, vous aurez vite des ennuis.

La mémoire de l'Apple II s'organise comme ceci :




← emplacement où vient se loger un programme BASIC.

les ennuis vont se traduire par l'apparition de tas de choses bizarres sur la page II. La raison est simple : le programme BASIC, après avoir envahi la page I (non utilisée pour le graphisme) se met à gignoter votre page 2.

Une solution consiste à mettre en tête de programme

$\emptyset$  LOMEM : 24576

Ceci a pour effet de tout reporter, programme, fichiers, tout ce que vous faites à la machine, au dessus de la page II :



↑  
↑  
↑  
← Début programme BASIC

C'est une bonne solution.

Mais vous ne pourrez plus, en utilisant **CHAIN**, conserver les fichiers

**PANGRAPHÉ DESSIN** est en fait bien loin d'être achevé, puisqu'il manque tout le traitement lignes vues, lignes cachées, ombages, etc ...

Suggestion : créer un troisième programme **PANGRAPHÉ DESSIN II** qui pourra être appellé par la première partie de **PANGRAPHÉ DESSIN**.


On peut ainsi, si on dispose de deux drives, manœuvrer in fine un outil dont l'énorme programme peut utiliser une part importante d'une disquette, tandis que l'objet, attaqué à travers le drive II pourra être constitué de milliers de segments.

au point de vue saisie vous pourrez gagner beaucoup de temps si vous disposez d'un digitaliseur (table graphique).

Avec une **TABLE TRAÇANTE** vous pourrez travailler très finement, au dixième de millimètre. Mais avec une simple imprimante faisant de la **RÉCOPIE D'ÉCRAN** vous pourrez conserver des copies papier de vos dessins.









A ce sujet, une astuce possible = au lieu de pointer vers le point G vous pouvez effectuer quatre dessins, au prix de quatre visées successives en

$G_1$   $G_2$   $G_3$   $G_4$ . Tirez les images sur papier, et assemblez. Vous quadruplez votre pouvoir de résolution. Et c'est comme si vous travailliez avec un Apple ayant une matrice graphique de 560 x 380 points ...












- Si j'en crois les chroniques de l'époque, l'auteur de ce programme me n'avait guère de crédits pour ses recherches.
- Ça devait être en tout cas un homme très soigneux, capable de rentrer toutes ces lignes programme bout à bout sans se tromper !...
- C'est assez étonnant, quand on voit la photo de son laboratoire ...
- Peut-être qu'il se trompait souvent, mais que c'était un homme d'un grand courage ...



Une place de  
village, avec  
son clocher, son  
marché, sa  
mairie, créée par  
**PANGRAPHÉ**.

Differentes vues  
extraites d'une  
animation.





# COMMENT UTILISER PANGRAPH

La disquette porte deux programmes :

PANGRAPH OBJET

PANGRAPH DESSIN


Vous pouvez lancer au départ l'un ou l'autre de ces programmes puisque le **CHAIN** et le sous-programme de transfert 40.000 permettra de travailler comme si c'était un programme unique. Nous allons commencer par **PANGRAPH OBJET**.

On fait donc RUN PANGRAPH OBJET

Puis on choisit l'option 1-CREER UN OBJET

et la sous-option 1.CREER DES CHAINES

Nous allons maintenant créer un carré situé comme ceci :



Le carré va être entré sous forme d'une chaîne constituée de 5 points

N°POINT	X	Y	Z
1	2	0	-4
2	9	0	-4
3	9	0	4
4	2	0	4
5	2	0	-4

Pas d'erreur, retour au menu.

Nous appelons alors l'option 9 - MANIPULER UN OBJET

Puis la sous option 2 - OPERER UNE ROTATION.

Apparaît :

L'OBJET EST-IL EN MEMOIRE ?

Réponse : oui

Nous allons alors choisir une rotation selon un axe parallèle à l'axe Oz ( choix 3 ).

L'idée est de faire tourner ce carré comme ceci de 90 degrés.

La trace de l'axe dans le plan XOY

correspond à  $X_A = 6 \quad Y_A = 0$

ANGLE, DEGRES : 90

Retour menu principal.


Cette fois-ci nous allons **STOCKER** cette objet en optant pour le choix 2. Et nous appellerons cet objet O1. La rotation du drive signale que ce fichier O1 est enregistré sur disque.

Retour menu, option 9 puis sous option

3 - CREER LE SYMETRIQUE % UN PLAN

L'objet O1 est toujours en mémoire.

Nous choisissons le plan YOZ, ce qui va substituer, en mémoire l'objet O2, symétrique de O1 :



Nous stockons ce nouvel objet O2.

Retour menu. Choix 9

Et sous option

4 - FUSIONNER OBJETS

O2 est toujours en mémoire.

La machine demande qu'on lui indique le nom de l'objet à fusionner. Nous choisirons O1.

Nous pourrions ainsi fusionner toute une série d'objets. Pour interrompre cette opération on tapera le mot FIN et nous appellerons l'objet global O


Cet objet est donc la réunion de O1 et O2.

Retour menu, option 9, sous option


## 2 - OPERER UNE ROTATION

L'objet est en mémoire et nous provoquons une rotation de  $90^\circ$  autour de l'axe Oz (sous-sous option 3,  $X_A = \emptyset$ ,  $Y_A = \emptyset$ )

Retour Menu. Stockage de cet objet qui on appellera O'.



Retour menu. Option 9. Sous option - L'objet O' est toujours en mémoire. Nous choisissons une rotation autour de l'axe OXT ( $YA = \emptyset$ ,  $Z_A = \emptyset$ , ANGLE, DEGRES :  $90^\circ$ ) on stocke cet objet O''.



Retour menu. Option 9

sous option 4 (fusion). O'' est toujours en mémoire.


On fusionne avec O' et O

et on nomme cet objet global **CUBE( $\emptyset/\emptyset/\emptyset$ )**

Entre parenthèses : les coordonnées du centre de gravité de cet objet.

Sur la disquette nous avons stocké un objet appelé  
**TRIEDRE( $\phi/\phi/\phi$ )**

Une flèche indique l'axe OXT.



Nous allons passer à la phase dessin et nous choisissons l'option 6-(REPRESENTER UN OBJET). Elle figure entre parenthèse dans le menu. On entendra donc l'ordinateur changer de programme, et chainer sur **PANGRAPHE DESSIN**

OUVERTURE ANGULAIRE : 2 $\phi$

POINT D'OBSERVATION

X = -25

Y = -8

Z = 12

CENTRE DE L'IMAGE

XG =  $\phi$


YG =  $\phi$

ZG =  $\phi$

L'objet **O** est en principe toujours en mémoire. Ce qui fait qu'en choisissant l'option

1- OBJET DEJA EN MEMOIRE

Nous voyons se dessiner :




Nous pouvons stocker cette image sur disque, et la rappeler, à volonté.

Retour menu Exercice : en gardant les mêmes paramètres de vision, tracez, par éléments

{ TRIEDRE (Ø/Ø/Ø)
{ Ø1
{ Ø2

on obtient l'image suivante :

Refaire l'opération en PAS à PAS.



Supposons que les objets  $O$ ,  $O'$ ,  $O''$  soient trop importants pour pouvoir être fusionnés (ce qui n'est pas le cas).

Nous allons, sur le Menu, choisir l'option

1Ø - DEFINIR UN BLOC D'OBJETS

NOMBRE D'OBJETS : 3

OBJET N° 1

$O$

OBJET N° 2

$O'$

OBJET N° 3

$O''$

NOM DE CE BLOC BLOC CUBE (Ø/Ø/Ø)

Nous pouvons alors envisager un dessin **PAR BLOC**

Passons (option 6) en Paragraphe dessin

mêmes paramètres de vision. Et nous choisissons la sous option 4 - DESSIN PAR BLOC


Les trois objets du bloc sont gérés automatiquement.

# LAVISION TOTALE

L'œil humain possède un certain **CHAMP VISUEL**. Certains animaux ont des champs visuels très supérieurs, par exemple les poissons. Ce qui a donné lieu à des objectifs photographiques correspondant à des ouvertures angulaires de  $90^\circ$  (**FISH-EYE** ou œil de poisson).


Si on a en mémoire un bloc grille composé d'éléments GRILLE 1, GRILLE 2, GRILLE 3, GRILLE 4, situés par exemple dans le plan  $Z = \phi$  on pourra observer cette **MIRE** en situant l'observateur sur l'axe OZT de telle manière que le point visé soit en  $(\phi, \phi, \phi)$ .

Voici l'image de cette mire, vue à une "altitude"  $Z = 1$  et avec une ouverture angulaire  $\Delta\theta = 90^\circ$



On retrouve l'aspect caractéristique des images FISH-EYE.


Ce qu'on pourra faire, maintenant, c'est construire un bloc suggérant une sorte de cage à lapins cubique, faite de grillage à maille carrée.



Il faudra partir de l'ensemble GRILLE 1, GRILLE 2, GRILLE 3, GRILLE 4, situé comme ceci :

Et utiliser les options rotation et translation.

Chaque face du cube sera constituée de 4 éléments, et le cube complet aura  $6 \times 4 = 24$  éléments.





On peut alors mettre la "mouche" dans la cage à lapins.  
Il suffira de donner une valeur AN=180°.

La mouche pourra voir tous les éléments de la cage à lapin d'un seul coup d'œil.



Sur l'écran, on pourrait tracer deux cercles concentriques. Le centre de l'écran correspond à l'axe de visée.

Un premier cercle, de rayon 42.5 correspond à AN=90°. C'est la limite du champ de vision FISHEYE.

Un deuxième cercle, de rayon 85, correspond au **POINT OCCIPITAL**, c'est à dire situé sur la direction diamétralement opposée à l'**AXE DE VISÉE**.



Sur la page suivant, la "CAGE"  
au centre de laquelle on va placer la 




Voici la structure de la face bouchée faisant partie de la CAGE. Elle est formée de quatre éléments "écornés" (voir page 50). Ainsi aucun des points de ce maillage n'appartient à l'axe OXT.

Il suffira de poser la mouche par exemple au point


$$(\phi; \phi; \phi)$$

et de la faire regarder dans la direction du point  $(\phi; \phi; \phi)$ .  
Valeur de AN =  $180^\circ$ .



Il faudra prendre la précaution de ne pas matérialiser le point occipital.




Dans ces condition, voila l'image de la cage cubique vu à travers cet **ŒIL DE MOUCHE**.




Sur le schéma ci-contre  
les six faces du cube. La  
sixième est évidemment un  
peu bizarre





Bon, maintenant nous savons comment voient les mouches, les poissons, et même les hommes...

- Nous avons compris comment ils voyaient dans leur tête le monde sensible, en fabriquant ces images sur leur rétine.
- les tableaux du musée sont donc la concrétisation de ce que les hommes voyaient dans leur tête.
- Faut croire...
- Bon, qui est-ce que c'est que ce truc là-bas ?
- L'étiquette dit que c'est l'Angélus de Millet.
- Tu sens quelque chose ?
- Rien. Au point de vue émotion artistique, c'est raté ...
- Peut-être qu'avec un 128 K ?...



**FIN**